

HISTORIK.BAS	0.66	18. 3.2017
--------------	------	------------

B E S C H R E I B U N G :

- Programm zur Erstellung historischer Zeittafeln

Q U E L L E N - A N G A B E N :

9. Gedenket des Vorigen von Alters her; denn ich bin Gott, und keiner mehr, ein Gott, desgleichen nirgend ist.

10. Der ich verkündige zuvor, was hernach kommen soll, und vorhin, ehe denn es geschieht, und sage: Mein Anschlag besteht, und ich thue Alles, was mir gefällt.

(Jesaja 46)

18. Denn ich sage euch, wahrlich, bis daß Himmel und Erde zergehe, wird nicht zergehen der kleinste Buchstabe, noch ein Titel vom Gesetz, bis daß es alles geschehe.

(Matthäus 5)

Bibel

"Die Bibel, oder die ganze Heilige Schrift Alten und Neuen Testaments" nach der deutschen Uebersetzung Dr. Martin Luthers.", Revision durch Dr. J. Ph. Fresenius, (1751); Druck und Verlag von Heinrich Ludwig Brönner, Frankfurt am Main, 40.Auflage, (1841)

[Rusk1974]

Rusk R. in "Christianity Today", Washington D.C., Vol XIII, Nr. 13, (29. 3.1974), pp 4 (720) - 6 (722);

Ohne Kommentar aus dem Englischen übersetzt von Erhard Kietz, Phys. i.R.

[Zem1987]

Zemanek H.: "Kalender und Chronologie", R. Oldenbourg Verlag München, Wien, 4. verbesserte Auflage, (1987)

D A N K :

Folgende weitere Personen haben zum Gelingen der Arbeit beigetragen:

- Gebhard Böhm, Pfarrer und Religionslehrer: Quellenanalyse 1986
- Joachim Wohlfahrt, Lehrer am Schubartgymnasium Aalen, 1986-1989
- Eckhard Walter, Stetten, Sinsheim und Adelshofen, ab 1989
- Markus Britsch, Pforzheim: Bergfreizeit 1992 mit erstem Rechenergebnis
- Herbert Henheik, Tübingen: Diskussion und Literatur, 1994
- Stefan Müller, Böblingen: Computer-Freizeit 1994 in Lausnitz
- Karl-Heinz Eismann, Lausnitz: Freizeit 1994 und Tagung 2000
- Georg Grittmann, Sinsheim: Publikation auf Computermisions-CD 2000
- Andreas Mulack, Langenorla: Tagung 2000 mit Vorführung des Algorithmus
- Gerd Baumann, Herrsching: Diskussion anhand Astronomie-Paket, ab 2001
- Claus-Edwin Jost, Schw. Gmünd: Prüfung der Datei HISTORIK.BAT, 2005
- Thomas Antoni, Erlangen: Englische Fassung von QBASIC, 2007
- Wolfgang Bay, Aalen: Hilfe bei der Auswahl englischer Bibeln, 2009
- Wilhelm Hecke, Ellwangen: Wiedergabe englischer Bibelzitate, 2009
- Johannes Stimmer, Aalen: Besorgung der DOS-Box 0.73 für Windows, 2009
- Camil Pogolski, Aalen: Besorgung der DOS-Box für Linux und Mac, 2009
- Heinz Döbele, Aalen: Chronologische Zusammenfassung der Bibel, 2016

B E A R B E I T U N G :

29.11.1994 - 19.12.1994	Norbert Südland, Aalen
30. 8.1999 - 27. 2.2000	Norbert Südland, Ulm/Donau
15. 7.2001 - 15. 9.2001	Norbert Südland, München
31. 8.2002 - 3.12.2007	Norbert Südland, Aalen
20. 1.2009 - 13. 8.2009	Norbert Südland, Aalen
18. 8.2012 - 18. 3.2017	Norbert Südland, Aalen

VORBEREITUNG

Erste Befehle:

```

OPTION BASE 1                'Felder beginnen beim Eintrag Nr.`1`!
'OPTION EXPLICIT
COMMON HistAuswahl%, Arbeitsplatz$, Zeit$, Zaehlweise%

```

Konstanten:

```

CONST VersionErstellDatum$ = "Version 0.66 vom 18. 3.2017"
CONST KonfigurationsDatei$ = "HISTORIK.CFG"

```

Unterprogramme anmelden:

Grundmenü:

```

DECLARE SUB Startbild (Antwort%, Eingabeerfolg%)
DECLARE SUB Pause ( )

DECLARE FUNCTION Abfrage$ (Text$, Vorgabe$, Eingabeerfolg%)
DECLARE FUNCTION Anlegen% (Daten$, Position&, HistAuswahl%, Eingabeerfolg%)
DECLARE FUNCTION Datieren% (HistAuswahl%, Auswahl&, Termin$)
DECLARE FUNCTION EinzelEingabe% (Daten$, z%, Farbe%, Eingabeerfolg%)
DECLARE FUNCTION Starttext$ (Antwort%)
DECLARE FUNCTION Taste$ ( )

```

Daten verändern:

```

DECLARE SUB Addieren (Summand1$, Art$, Richtung$, v$, OZahl1%, OZahl2%, W$)
DECLARE SUB Aendern (Daten$, Variable$, Inhalt$)

```

```
DECLARE SUB Schreiben (Datei%, DatensatzLaenge%, Position&, Daten$)
DECLARE SUB StatusNeu (Datum$)
```

```
DECLARE FUNCTION Editor$ (Vorgabe$, Laenge%, Modus%, Eingabeerfolg%)
DECLARE FUNCTION FuerDatei$ (Symbol$)
DECLARE FUNCTION FuerNutzer$ (Symbol$)
DECLARE FUNCTION GleichzeitigTest$ (Text$)
DECLARE FUNCTION TextAuffuellen$ (Text$, WunschLaenge%, Zeichen$)
```

```
'-----'
'|                                     |'
'|                               Daten anzeigen:                               |'
'|-----'
```

```
DECLARE SUB Ausgabe (Farbe%, Frage$, Puffer$, Offs%, a$, b%, Bereiche%, y%)
DECLARE SUB Datumsanzeige (Daten$, z%, Farbe%, HistAuswahl%)
DECLARE SUB LINEINPUT (Datei%, Zeile$)
DECLARE SUB ZeilenAusgabe (Daten$, z%, Farbe%)
```

```
DECLARE FUNCTION AktuellesDatum$ ()
DECLARE FUNCTION Namenseintrag$ (Position&)
DECLARE FUNCTION Ueberblick& (HistAuswahl%, Vorgabe&, Eingabeerfolg%)
DECLARE FUNCTION Ereignis$ (Symbol$)
DECLARE FUNCTION Laden$ (Datei%, DatensatzLaenge%, Position&)
DECLARE FUNCTION Nummer% (Text$, Start%, Ende%)
DECLARE FUNCTION SIZEOF% (Text$)
DECLARE FUNCTION STRLEN% (Text$, Ende$)
DECLARE FUNCTION Teil$ (Daten$, Variable$)
DECLARE FUNCTION Wochentag% (jh%, j%, m%, t%)
```

```
'-----'
'|                                     |'
'|                               Umrechnungen:                               |'
'|-----'
```

```
DECLARE SUB Restklasse (Vorzeichen%, Jahr%, Monat%, Tag%)
DECLARE SUB TerminZuZahl (Termin$, Vorzeichen%, Jahr%, Monat%, Tag%)
DECLARE SUB Umkehren (v1%, j1%, m1%, t1%, v2%, j2%, m2%, t2%, Richtung$)
DECLARE SUB ZeitSchieben (v1%, j1%, m1%, t1%, S$, v2%, j2%, m2%, t2%, R$)
```

```
DECLARE FUNCTION ZahlZuTermin$ (VZ%, Jahr%, Monat%, Tag%)
```

```
DEF FNMAX (a, b) = (a + b) / 2 + ABS(a - b) / 2
DEF FNMIN (a, b) = (a + b) / 2 - ABS(a - b) / 2
```

```
'-----'
'|                                     |'
'|                               Globale Variablen:                               |'
'|-----'
```

```
'Stack-Größe festlegen:
```

```
'-----'
```

```
IF Zaehlweise% = 0 THEN
  CLEAR , , 4096
END IF
```

```
'DIM AS STRING:
```

```
'-----'
```

```

DIM SHARED LeerZeichen$
DIM SHARED GlobDatum$

'DIM AS INTEGER:
'-----'
DIM SHARED GlobWochentag%
DIM SHARED GlobVarAnzahl% 'Variablen-Zahl der Struktur-Beschreibung
DIM SHARED Eingabe%       'Referenz-Nummer der Eingabe-Datei
DIM SHARED DatenLaenge%   'Datenlänge einer Eingabe-Zeile

'DIM AS LONG:
'-----'
DIM SHARED GEL&           'Zeilen-Anzahl der Eingabe-Datei

'Dynamische Datenfelder:
'-----'
REDIM SHARED GVAufang%(1) 'Variablen-Anfang für MID$-Zugriff
REDIM SHARED GVLaege%(1)  'Variablen-Laenge für MID$-Zugriff
REDIM SHARED GVName$(1)   'Variablen-Name für Teil$-Zugriff
REDIM SHARED GVTyp$(1)    'Variablen-Typ für Teil$-Zugriff

```

```

'-----'
' Lokale Variablen:
'-----'

```

```

'-----'
' Vorbereitung:
'-----'

```

```

'DIM AS STRING:
'-----'
DIM Jahrhundert$
DIM EingabeDatei$
DIM HistorikAuswahl$
DIM t$           'Tag
DIM m$           'Monat
DIM j$           'Jahr
DIM Jahr$
DIM Text$        'Datenpuffer
DIM erg$
DIM Datum$
DIM Daten$
DIM Tausch$
DIM Vergleich$
DIM Puffer$
DIM Zeile$
DIM Variable$
DIM Summand1$
DIM Summand2$
DIM Differenz$
DIM Termin$
DIM Status$
DIM Gleichzeitig$
DIM KopieDatei$
DIM PositionsDatei$
DIM Antwort$
DIM Vorgabe$

```

```

'DIM AS INTEGER:
'-----'
DIM Konfiguration%
DIM Leseversuch%
DIM Position%
DIM Laenge%
DIM jh%           'Jahrhundert
DIM j%           'Jahr
DIM Jahr%
DIM m%           'Monat
DIM t%           'Tag
DIM x%           'Spaltenzähler
DIM y%           'Zeilenzähler
DIM Eingabeerfolg%
DIM Erfolg%
DIM Halt%
DIM OK%
DIM Neu%
DIM Geaendert%
DIM Anders%
DIM Ausstieg%
DIM Kopie%
DIM Offset%
DIM DirDat%
DIM HQLDat%       'HQL = Historische Quellenlage

'DIM AS LONG INTEGER:
'-----'
DIM z&           'Zähler
DIM Auswahl&
DIM Vorgabe&
DIM Ziel&
DIM Anzahl&

```

HAUPTTEIL

ON ERROR GOTO FehlerBehandlung

```

'Auswahlmenü:
'-----'
GOSUB W0      'Vorbereitung
DO
    Eingabeerfolg% = 1
    Startbild HistAuswahl%, Eingabeerfolg%
    IF Eingabeerfolg% = 1 THEN
        ON HistAuswahl% GOSUB W1, W2, W3, W4, W5, W6, W7, W8, W9
    END IF
LOOP UNTIL HistAuswahl% = 0  'Explizite Programmbeendigung nötig!

'=====
Programmende:
'=====
    GOSUB KonfigurationAbspeichern
SYSTEM

```

' _____ ENDE DES HAUPTTEILS _____ '

FEHLERBEHANDLUNG

'====='

FehlerBehandlung:

'====='

```

SELECT CASE ERR
CASE 5                'unzulässiger Funktionsaufruf
    SELECT CASE ERL
    CASE 12, 13, 14, 15
        RESUME 10
    CASE 107
        RESUME 109
    CASE ELSE
        Pause
        RESUME NEXT
    END SELECT
CASE 7                'zu wenig Speicher
    SELECT CASE ERL
    CASE 5                'Umgebungsvariable setzen
        RESUME NEXT
    CASE 11
        Text$ = "Bitte das aktuelle Jahrhundert mit dem DOS-Befehl " + CHR$(34)
        Text$ = Text$ + "SET CENTURY=" + Jahrhundert$ + CHR$(34) + " angeben."
        Ausgabe 1, Text$, "", 0, "M", 1, 1, 9
        Pause
        RESUME NEXT
    CASE ELSE
        Pause
        RESUME NEXT
    END SELECT
CASE 9                'Index außerhalb des zulässigen Bereichs
    SELECT CASE ERL
    CASE 801            'Funktion `SIZEOF%()`
        PRINT "in der Positionsdatei "; CHR$(34); PositionsDatei$; CHR$(34);
        PRINT "."
        PRINT
        COLOR 0, 7
        PRINT " Das Programm wird abgebrochen! ";
        COLOR 7, 0
        Pause
        GOTO Programmende
    CASE ELSE
        Pause
        RESUME NEXT
    END SELECT
CASE 52                'Dateiname oder -nummer unzulässig
    SELECT CASE ERL
    CASE 3, 4            'Konfigurationsdatei einlesen
        RESUME NEXT
    CASE ELSE
        Pause
        RESUME NEXT
    END SELECT

```

```
CASE 53          'Datei nicht gefunden
SELECT CASE ERL
CASE 2           'Konfigurationsdatei öffnen
  PRINT "Konfigurationsdatei "; CHR$(34); KonfigurationsDatei$; CHR$(34)
  PRINT "fehlt und wird erstellt."
  RESUME NEXT
CASE 6
  PRINT "Strukturdatei "; CHR$(34); "HISTORIK.STR"; CHR$(34); " fehlt."
  PRINT
  COLOR 0, 7
  PRINT " Bitte das Programm über HISTORIK.BAT starten! "
  COLOR 7, 0
  Pause
  GOTO Programmende
CASE 7           'Positionsdatei vorbereiten
  PRINT "Programmteil "; CHR$(34); "STRUKTUR.BAS"; CHR$(34); " fehlt."
  PRINT
  COLOR 0, 7
  PRINT " Bitte das Programm über HISTORIK.BAT starten! "
  COLOR 7, 0
  Pause
  GOTO Programmende
CASE 8           'Positionsdatei öffnen
  PRINT "Positionsdatei "; CHR$(34); PositionsDatei$; CHR$(34)
  PRINT "fehlt und wird erstellt."
  Leseversuch% = Leseversuch% + 1
  RESUME 6
CASE 61
  PRINT
  PRINT "Programmteil "; CHR$(34); "RECHNEN.BAS"; CHR$(34); " fehlt."
  Pause
  RESUME NEXT
CASE 92
  PRINT
  PRINT "Programmteil "; CHR$(34); "DRUCKEN.BAS"; CHR$(34); " fehlt."
  Pause
  GOTO Programmende
CASE 101         'Öffnen von `Arbeitsplatz$ + EingabeDatei$`
  RESUME 105     '`EingabeDatei$` nach `Arbeitsplatz$` kopieren.
CASE ELSE
  Pause
  RESUME NEXT
END SELECT
CASE 62         'Einlesen nach Dateiende
SELECT CASE ERL
CASE 3
  PRINT "Leere Konfigurationsdatei ";
  PRINT CHR$(34); KonfigurationsDatei$; CHR$(34); "."
  PRINT "Bitte laden Sie die ursprüngliche Datei-Umgebung in das ";
  PRINT "aktuelle Verzeichnis."
  Pause
  GOTO Programmende
CASE 4
  RESUME NEXT
CASE 8
  PRINT "Leere Positionsdatei "; CHR$(34); PositionsDatei$; CHR$(34); "."
  Pause
  GOTO Programmende
```

```
CASE 9
  PRINT "Unvollständige Positionsdatei ";
  PRINT CHR$(34); PositionsDatei$; CHR$(34); "."
  Pause
  GOTO Programmende
CASE 100
  Vorgabe& = 1
  Zeile$ = Puffer$
  RESUME NEXT
CASE ELSE
  Pause
  RESUME NEXT
END SELECT
CASE 70                                'Zugriff verweigert
  SELECT CASE ERL
    CASE 1                            'Schreibtests auf `Arbeitsplatz$`
      PRINT "Das Arbeitsplatz-Verzeichnis " + Arbeitsplatz$
      PRINT "ist schreibgeschützt. Es wird über " + CHR$(34);
      PRINT "SET HISTORIKTEMP=.." + CHR$(34) + " auf DOS-Ebene gesetzt."
      Pause
      SYSTEM                          'Harter Programm-Abbruch
    CASE ELSE
      Pause
      RESUME NEXT
  END SELECT
CASE 75                                'Pfad- /Datei-Zugriffsfehler
  SELECT CASE ERL
    CASE 1                            'Schreibtests auf `Arbeitsplatz$`
      RESUME NEXT
    CASE 102                          'Schreibschutz auf `Eingabedatei$`
      RESUME 104
    CASE ELSE
      Pause
      RESUME NEXT
  END SELECT
CASE 76                                'Pfad nicht gefunden
  SELECT CASE ERL
    CASE 1
      PRINT "Das gewählte Arbeitsverzeichnis " + Arbeitsplatz$
      PRINT "wurde nicht gefunden. Das Programm wird beendet."
      Pause
      SYSTEM
    CASE ELSE
      Pause
      RESUME NEXT
  END SELECT
CASE 100
  PRINT "Datenstruktur fehlt in "; CHR$(34); "HISTORIK.POS"; CHR$(34); "."
  Pause
  GOTO Programmende
CASE 106
  PRINT "Zu große Jahreszahl. 4-stellig sollte reichen.";
  Pause
  RESUME NEXT
CASE 110
  PRINT "Sinnlose Datierung mit Ordnungszahl-Chaos."
  Pause
  GOTO Programmende
```

```

CASE 111
  PRINT "Unbekannte Struktur kann nicht verändert werden."
  Pause
  GOTO Programmende
END SELECT
ON ERROR GOTO 0
GOTO Programmende
' _____ ENDE DER FEHLERBEHANDLUNG _____ '

'
' _____
' SUBROUTINEN VIA GOSUB
' _____
'

'=====
W0:      'Vorbereitung:
'=====
' Ermittelt alle Angaben für das Hauptmenü.
'
' Bearbeitung:
' 18. 8.2001 - 19. 8.2001: Norbert Südland, Adelshofen
' 31. 8.2002 - 6. 9.2002: Norbert Südland, Aalen
' 22.11.2006 - 4. 5.2007: Norbert Südland, Aalen
' Überprüfung:
' 31. 8.2002 - 6. 9.2002: Norbert Südland, Aalen
'-----
SCREEN 0, 0
COLOR 7, 0

' Als `LeerZeichen$` kommt prinzipiell `CHR$(32)` und `CHR$(255)` in
' Frage, wodurch die folgende Festlegung generell Vorteile gegenüber
' einer Konstante `LeerZeichen$ = " "` besitzt:
'-----
LeerZeichen$ = SPACE$(1)

' Als `Arbeitsplatz$` wird das Arbeitsverzeichnis verwendet, das
' %HISTORIKTEMP%, %TEMP% oder %TMP% angibt.
' Falls (bei alten DOS-Versionen) gar kein `Arbeitsplatz$` angegeben ist,
' wird versucht, auf dem Datenträger, der auch das Programm enthält, zu
' schreiben. Eventuell bricht das Programm ab, wenn der `Arbeitsplatz$`
' nicht beschreibbar ist:
'-----
Arbeitsplatz$ = ENVIRON$( "HISTORIKTEMP" )
IF Arbeitsplatz$ = "" THEN
  Arbeitsplatz$ = ENVIRON$( "QBASICTEMP" )
  IF Arbeitsplatz$ <> "" THEN Arbeitsplatz$ = Arbeitsplatz$ + "\"
1  MKDIR Arbeitsplatz$ + "HISTORIK.TMP"
  Arbeitsplatz$ = Arbeitsplatz$ + "HISTORIK.TMP\"
ELSE
  IF Arbeitsplatz$ <> "" THEN Arbeitsplatz$ = Arbeitsplatz$ + "\"
END IF

'Beschreibbarkeit von `Arbeitsplatz$` prüfen:
'-----
BSAVE Arbeitsplatz$ + "HISTORIK.CHK", 0, 0
KILL Arbeitsplatz$ + "HISTORIK.CHK"

'Konfigurationsdatei öffnen:

```

```

'-----'
Eingabeerfolg% = 0
WHILE Eingabeerfolg% < 2 AND HistorikAuswahl$ = ""
    Konfiguration% = FREEFILE
    IF Eingabeerfolg% = 0 THEN
        Puffer$ = Arbeitsplatz$ + KonfigurationsDatei$
    ELSE
        Puffer$ = KonfigurationsDatei$
    END IF
2    OPEN Puffer$ FOR INPUT AS #Konfiguration%

    '`Jahrhundert$` einlesen:
    '-----'
3    LINE INPUT #Konfiguration%, Jahrhundert$

    'Weitere Konfigurationen einlesen:
    '-----'
4    LINE INPUT #Konfiguration%, EingabeDatei$
    LINE INPUT #Konfiguration%, HistorikAuswahl$
    LINE INPUT #Konfiguration%, Text$
    IF Zaehlweise% < 1 OR Zaehlweise% > 4 THEN
        Zaehlweise% = VAL(Text$)
    END IF
    CLOSE #Konfiguration%
    Eingabeerfolg% = Eingabeerfolg% + 1
WEND

'Prüfen, aktualisieren oder Vorgabewerte (default) setzen:
'-----'
jh% = VAL(RIGHT$(Jahrhundert$, 3))
IF jh% > 100 OR jh% < -10 THEN jh% = 0
IF jh% = 0 OR LEN(ENVIRON$("CENTURY")) = 4 THEN
    jh% = VAL(LEFT$(ENVIRON$("CENTURY"), 1) + "1")
    jh% = jh% * VAL(RIGHT$(ENVIRON$("CENTURY"), 3))
    IF jh% > 100 OR jh% < -10 OR jh% = 0 THEN
        jh% = 21          'Das 0.Jahrhundert existiert nicht!
    END IF
END IF
IF Zaehlweise% < 1 OR Zaehlweise% > 4 THEN
    Zaehlweise% = 1
END IF

'Besonderheiten von WindowsNT beachten:
'-----'
Jahrhundert$ = TextAuffuellen$(STR$(jh% * SGN(jh%)), -3, LeerZeichen$)
IF SGN(jh%) = -1 THEN
    Jahrhundert$ = "-" + Jahrhundert$
ELSE
    Jahrhundert$ = "+" + Jahrhundert$
END IF
5    ENVIRON "CENTURY=" + Jahrhundert$    'Versuch der Aktualisierung

EingabeDatei$ = UCASE$(EingabeDatei$)
EingabeDatei$ = LEFT$(EingabeDatei$, STRLEN$(EingabeDatei$, "."))
IF EingabeDatei$ = "" THEN EingabeDatei$ = "TEST"
EingabeDatei$ = TextAuffuellen$(EingabeDatei$, 8, LeerZeichen$) + ".HQL"

HistAuswahl% = VAL(HistorikAuswahl$)

```

```

IF HistAuswahl% < 1 OR HistAuswahl% > 9 THEN
    HistAuswahl% = 1
END IF

GOSUB KonfigurationAbspeichern

'Wochentag ausrechnen:
'-----'
jh% = VAL(Jahrhundert$)
j% = VAL(RIGHT$(DATE$, 2))
IF j% = 0 THEN
    j% = 100          'Wertebereich für 'j%': 1 bis 100
END IF
IF jh% = 100 AND j% = 100 THEN jh% = 96      'gleiche Wochentage
IF jh% = -10 AND j% = 1 THEN jh% = -3       'zum Datum
m% = VAL(LEFT$(DATE$, 2))
t% = VAL(MID$(DATE$, 4, 2))
GlobWochentag% = Wochentag%(jh%, j%, m%, t%)
t$ = TextAuffuellen$(LTRIM$(STR$(t%)), -2, LeerZeichen$)
m$ = TextAuffuellen$(LTRIM$(STR$(m%)), -2, LeerZeichen$)
j$ = TextAuffuellen$(LTRIM$(STR$(j%)), -4, LeerZeichen$)
GlobDatum$ = t$ + "." + m$ + "." + j$

PRINT "Strukturpositionen von HISTORIK.POS einlesen..."
Leseversuch% = 1
6  SELECT CASE Leseversuch%
CASE 1
    PositionsDatei$ = "HISTORIK.POS"
CASE 2
    PositionsDatei$ = Arbeitsplatz$ + PositionsDatei$
CASE 3
    Konfiguration% = FREEFILE
    OPEN "HISTORIK.STR" FOR INPUT AS #Konfiguration%
    CLOSE #Konfiguration%
7  OPEN Arbeitsplatz$ + "STRUKTUR.CFG" FOR OUTPUT AS #Konfiguration%
    PRINT #Konfiguration%, "HISTORIK.STR"      'Strukturdatei
    PRINT #Konfiguration%, PositionsDatei$    'tatsächliche Positionsdatei
    PRINT #Konfiguration%, "HISTORIK.BAS"     'Rücksprungprogramm
    CLOSE #Konfiguration%
    CHAIN "STRUKTUR.BAS"
CASE ELSE
    Pause                                     'Programmierfehler!
END SELECT

Position% = FREEFILE
8  OPEN PositionsDatei$ FOR INPUT AS #Position%
9  LINE INPUT #Position%, Text$
    Laenge% = STRLEN$(Text$, LeerZeichen$)
    GlobVarAnzahl% = VAL(LEFT$(Text$, Laenge%))
    PRINT LTRIM$(STR$(GlobVarAnzahl%)); " Strukturelemente"
    REDIM SHARED GVAnfang%(1 TO GlobVarAnzahl%)      'AS INTEGER
    REDIM SHARED GVLaeenge%(1 TO GlobVarAnzahl%)     'AS INTEGER
    REDIM SHARED GVName$(1 TO GlobVarAnzahl%)        'AS STRING
    REDIM SHARED GVTyp$(1 TO GlobVarAnzahl%)         'AS STRING
    FOR z% = 1 TO GlobVarAnzahl%
        INPUT #Position%, GVAnfang%(z%)
        INPUT #Position%, GVLaeenge%(z%)
        INPUT #Position%, GVName$(z%)

```

```

        INPUT #Position%, GVTyp$(z&)
    NEXT z&
CLOSE Position%

DatenLaenge% = SIZEOF%("Daten")
IF DatenLaenge% <= 0 THEN ERROR 100

RETURN 'W0 Vorbereitung _____'

'=====
KonfigurationAbspeichern:
'=====

'Aktuelle Konfiguration abspeichern:
'-----'
Konfiguration% = FREEFILE
OPEN Arbeitsplatz$ + KonfigurationsDatei$ FOR OUTPUT AS #Konfiguration%
PRINT #Konfiguration%, Jahrhundert$
PRINT #Konfiguration%, UCASE$(EingabeDatei$)
PRINT #Konfiguration%, HistAuswahl%
PRINT #Konfiguration%, Zaehlweise%
CLOSE #Konfiguration%

RETURN 'KonfigurationAbspeichern _____'

'=====
W1:      'Aktuelles Datum ändern:
'=====
' Ändert das "aktuelle" Datum unter Berücksichtigung der historischen
' Kalenderumstellungen am 29. 2.45 v.Chr. und am 15.10.1582.
' Es wird demonstriert, daß das Jahr-2000-Problem viel umfassender bereits
' mit DOS und BASIC gelöst werden kann.
'
' Bearbeitung:
' 6. 9.2002: Norbert Südland, Aalen
' 22.11.2006: Norbert Südland, Aalen
' Überprüfung:
'-----'

t$ = LEFT$(GlobDatum$, 2)
m$ = MID$(GlobDatum$, 4, 2)
Jahr$ = RIGHT$(GlobDatum$, 4)
Laenge% = LEN(AktuellesDatum$)
y% = INT(40! - Laenge% / 2) + Laenge% - 9
10 DO
    LOCATE 8, y%
    erg$ = Editor$(t$, 2, 1, Eingabeerfolg%)
    IF Eingabeerfolg% = 1 THEN
        t% = VAL(erg$)
    END IF
    LOOP WHILE (t% < 1 OR t% > 31) AND Eingabeerfolg% = 1
    IF Eingabeerfolg% = 0 THEN
        GOTO DatumSetzen
    END IF
    t$ = erg$
    PRINT t$;
    DO
        LOCATE 8, y% + 3
        erg$ = Editor$(m$, 2, 1, Eingabeerfolg%)

```

```

    IF Eingabeerfolg% = 1 THEN
        m% = VAL(erg$)
    END IF
    LOOP WHILE (m% < 1 OR m% > 12) AND Eingabeerfolg% = 1
    IF Eingabeerfolg% = 0 THEN
        GOTO DatumSetzen
    END IF
    m$ = erg$
    PRINT m$;
    DO
        LOCATE 8, y% + 6
        erg$ = Editor$(Jahr$, 4, 1, Eingabeerfolg%)
        IF Eingabeerfolg% = 1 THEN
            Jahr% = VAL(erg$)
        END IF
    LOOP UNTIL Jahr% <> 0 OR Eingabeerfolg% = 0
    IF Eingabeerfolg% = 0 THEN
        GOTO DatumSetzen
    END IF
    Jahr$ = erg$
    PRINT Jahr$;
'=====
DatumSetzen:
'=====
    Jahr% = VAL(Jahr$)
    IF Jahr% > 0 THEN
        jh% = INT((Jahr% - 1) / 100) + 1      'Wertebereich > 0
        j% = ((Jahr% - 1) MOD 100) + 1      'Wertebereich: 1-100
    ELSEIF Jahr% < 0 THEN
        jh% = INT(Jahr% / 100)              'Wertebereich < 0
        j% = 1 + Jahr% - 100 * jh%         'Wertebereich: 1-100
    END IF

    'Gregorianische Kalenderreform:
    '-----'
    Jahr% = VAL(Jahr$)
    IF t% = 29 AND m% = 2 AND Jahr% > 1582 AND Jahr% MOD 100 = 0 THEN
12     IF Jahr% MOD 400 <> 0 THEN ERROR 5
    END IF
13     IF Jahr% = 1582 AND m% = 10 AND t% < 15 AND t% > 4 THEN ERROR 5

    'Julianischer Kalender:
    '-----'
14     IF t% = 29 AND m% = 2 AND Jahr% < -45 THEN ERROR 5
        j$ = RIGHT$(STR$(j% + 100), 2)
        IF j% MOD 100 < 80 THEN
            Datum$ = LTRIM$(m$) + "/" + LTRIM$(t$) + "/20" + j$
        ELSE
            Datum$ = LTRIM$(m$) + "/" + LTRIM$(t$) + "/19" + j$
        END IF
15     DATE$ = Datum$
        GlobDatum$ = t$ + "." + m$ + "." + Jahr$

    'Wochentag ausrechnen und anzeigen (Funktion verändert j%):
    '-----'
    GlobWochentag% = Wochentag%(jh%, j%, m%, t%)

    LOCATE 8, 1, 0

```

```

PRINT SPACE$(80);
Ausgabe 1, AktuellesDatum$, "", 0, "M", 1, 1, 8

'aktuelles Jahrhundert in den Umgebungsvariablen prüfen:
'-----'
Jahrhundert$ = TextAuffuellen$(STR$(jh% * SGN(jh%)), -3, LeerZeichen$)
IF SGN(jh%) = -1 THEN
    Jahrhundert$ = "-" + Jahrhundert$
ELSE
    Jahrhundert$ = "+" + Jahrhundert$
END IF
11 IF ENVIRON$("CENTURY") <> Jahrhundert$ THEN ERROR 7
IF LEN(ENVIRON$("CENTURY")) = 4 THEN ENVIRON "CENTURY=" + Jahrhundert$

'aktuelles Jahrhundert in die Konfiguration abspeichern:
'-----'
GOSUB KonfigurationAbspeichern
RETURN 'W1 Aktuelles Datum ändern _____'

'=====
W2:      'Datei anlegen /erweitern:
'=====
GOSUB Dateiname
IF Erfolg% = 0 THEN RETURN
Halt% = 0
Eingabeerfolg% = 1
WHILE GEL& < 16777215 AND Halt% = 0 AND Eingabeerfolg% = 1
    GEL& = GEL& + 1
    Daten$ = SPACE$(SIZEOF$("Daten") - 3) + "|" + CHR$(13) + CHR$(10)
    OK% = Anlegen%(Daten$, GEL&, HistAuswahl%, Eingabeerfolg%)
    IF Eingabeerfolg% = 0 THEN
        GEL& = GEL& - 1
    ELSE
        IF OK% = 0 THEN
            GEL& = GEL& - 1
        ELSE
            Schreiben Eingabe%, DatenLaenge%, GEL&, Daten$
        END IF
    END IF
    IF Abfrage$("Datei beenden?", "N", Eingabeerfolg%) = "J" THEN
        IF Eingabeerfolg% = 1 THEN
            Halt% = 1
        END IF
    ELSE
        Halt% = 0
    END IF
WEND
CLOSE #Eingabe%
RETURN 'W2 Datei anlegen /erweitern _____'

'=====
W3:      'Eingabe löschen:
'=====
GOSUB Dateiname
IF Erfolg% = 0 THEN RETURN
DO
    Auswahl& = Ueberblick&(HistAuswahl%, Auswahl&, Eingabeerfolg%)
    IF Eingabeerfolg% = 1 THEN GOSUB Loeschen

```

```

    LOOP UNTIL Eingabeerfolg% = 0
    CLOSE #Eingabe%
RETURN 'W3 Eingabe löschen _____'

'=====
W4:      'Eingabe vordatieren:
'=====
    GOSUB Dateiname
    IF Erfolg% = 0 THEN RETURN
    DO
        Auswahl& = Ueberblick&(HistAuswahl%, Auswahl&, Eingabeerfolg%)
        IF Eingabeerfolg% > 0 THEN
            z& = Datieren%(HistAuswahl%, Auswahl&, Termin$)
        END IF
        LOOP UNTIL Eingabeerfolg% = 0
    CLOSE #Eingabe%
RETURN 'W4 Eingabe vordatieren _____'

'=====
W5:      'Übersicht und Korrektur:
'=====
    GOSUB Dateiname
    IF Erfolg% = 0 THEN RETURN
    Eingabeerfolg% = 1
    DO
        Auswahl& = Ueberblick&(HistAuswahl%, Auswahl&, Eingabeerfolg%)
        IF Eingabeerfolg% > 0 THEN
            IF Abfrage$("Vordatierungen ansehen?", "N", Eingabeerfolg%) = "J" THEN
                IF Eingabeerfolg% = 1 THEN
                    z& = Datieren%(HistAuswahl%, Auswahl&, Termin$)
                END IF
            ELSE
                IF Eingabeerfolg% = 1 THEN
                    Daten$ = Laden$(Eingabe%, DatenLaenge%, Auswahl&)
                    OK% = Anlegen%(Daten$, Auswahl&, HistAuswahl%, Eingabeerfolg%)
                    IF Eingabeerfolg% = 1 THEN
                        IF OK% = 0 THEN
                            GOSUB Loeschen
                        ELSE
                            Aendern Daten$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
                            Schreiben Eingabe%, DatenLaenge%, Auswahl&, Daten$
                        END IF
                    END IF
                END IF
            END IF
            Eingabeerfolg% = 1      'Zu starken Rücksprung bei [ESC] vermeiden.
        END IF
    LOOP UNTIL Eingabeerfolg% = 0
    CLOSE #Eingabe%
RETURN 'W5 Übersicht und Korrektur _____'

'=====
W6:      'Berechnung:
'=====
    GOSUB Dateiname
    IF Erfolg% = 0 THEN RETURN
    IF GEL& > 0 THEN
        Neu% = 0

```

```

IF Abfrage$( "Neuberechnung aller Daten?", "N", Erfolg%) = "J" THEN
  IF Erfolg% = 1 THEN Neu% = 1
END IF
IF Erfolg% = 1 THEN
  LOCATE 25, 1: PRINT SPACE$(79);
  Text$ = "Welche Rechenmethode? [1-4]"
  IF Zaehlweise% >= 1 AND Zaehlweise% <= 4 THEN
    Vorgabe$ = LTRIM$(STR$(Zaehlweise%))
  ELSE
    Vorgabe$ = "1"
  END IF
  Ausgabe 2, Text$, Vorgabe$, -1, "M", 1, 1, 25
  x% = CSRLIN
  y% = POS(0) - 1
  DO
    LOCATE x%, y%
    Zaehlweise% = VAL(Editor$(Vorgabe$, 1, 2, Erfolg%))
  LOOP UNTIL Zaehlweise% >= 1 AND Zaehlweise% <= 4 OR Erfolg% = 0
  IF Erfolg% = 1 THEN
    PRINT LTRIM$(STR$(Zaehlweise%));
  END IF
END IF
IF Erfolg% = 1 THEN
  Konfiguration% = FREEFILE
  OPEN Arbeitsplatz$ + "RECHNEN.CFG" FOR OUTPUT AS #Konfiguration%
  PRINT #Konfiguration%, UCASE$(EingabeDatei$)
  PRINT #Konfiguration%, GEL&
  PRINT #Konfiguration%, Neu%
  PRINT #Konfiguration%, Zaehlweise%
  PRINT #Konfiguration%, "HISTORIK.BAS "           'Rücksprungprogramm
  CLOSE #Konfiguration%
END IF
END IF
CLOSE #Eingabe%
IF GEL& > 0 AND Erfolg% = 1 THEN
  GOSUB KonfigurationAbspeichern
61 CHAIN "RECHNEN.BAS "
END IF
RETURN 'W6 Berechnung _____

'=====
W7:      'Datei sortieren:
'=====
' Ermöglicht ein manuelles Sortieren im Sinne eines Kartei-Kastens:
'
' Bearbeitung:
' 18.12.2002: Norbert Südland, Aalen
'-----
GOSUB Dateiname
IF Erfolg% = 0 THEN RETURN
Eingabeerfolg% = 1
DO
  Auswahl& = Ueberblick&(6, Auswahl&, Eingabeerfolg%)
  IF Eingabeerfolg% > 0 THEN
    Daten$ = Laden$(Eingabe%, DatenLaenge%, Auswahl&)
    Ziel& = Ueberblick&(7, Auswahl&, Eingabeerfolg%)
    IF Eingabeerfolg% > 0 THEN
      IF Auswahl& <= Ziel& THEN

```

```

        FOR z& = Auswahl& TO Ziel& - 1&
            Tausch$ = Laden$(Eingabe%, DatenLaenge%, z& + 1)
            Aendern Tausch$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
            Schreiben Eingabe%, DatenLaenge%, z&, Tausch$
        NEXT z&
    ELSE
        FOR z& = Auswahl& TO Ziel& + 1& STEP -1
            Tausch$ = Laden$(Eingabe%, DatenLaenge%, z& - 1)
            Aendern Tausch$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
            Schreiben Eingabe%, DatenLaenge%, z&, Tausch$
        NEXT z&
    END IF
    Aendern Daten$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
    Schreiben Eingabe%, DatenLaenge%, Ziel&, Daten$
END IF
END IF
LOOP UNTIL Eingabebeerfolg% = 0
CLOSE #Eingabe%
RETURN 'W7 Datei sortieren _____'

'=====
W8:      'Datei umdatieren:
'=====
' Letzte Korrektur:
'   9.10.2007 Norbert Südland
'-----'
GOSUB Dateiname
IF Erfolg% = 0 THEN RETURN
DO
    Auswahl& = Ueberblick&(8, Auswahl&, Eingabebeerfolg%)
    IF Eingabebeerfolg% = 0 THEN
        Geaendert% = 0
    ELSE
        Daten$ = Laden$(Eingabe%, DatenLaenge%, Auswahl&)
        Anders% = Datieren%(HistAuswahl%, Auswahl&, Termin$)
    END IF
LOOP UNTIL Anders% <> 0 OR Eingabebeerfolg% = 0
IF LTRIM$(Termin$) <> "" THEN
    Vergleich$ = Laden$(Eingabe%, DatenLaenge%, Auswahl&)
    Aendern Vergleich$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
    Schreiben Eingabe%, DatenLaenge%, Auswahl&, Vergleich$

    Variable$ = "Daten.Datum[" + LTRIM$(STR$(INT((Anders% + 1) / 2))) + "]"
    Summand1$ = Teil$(Vergleich$, Variable$)
    IF Anders% MOD 2 = 1 THEN
        Aendern Summand1$, "Zeitangabe.Minimum", Termin$
    ELSE
        Aendern Summand1$, "Zeitangabe.Maximum", Termin$
    END IF
    Aendern Summand1$, "Zeitangabe.Status", SPACE$(1)
    StatusNeu Summand1$
    Summand2$ = Teil$(Daten$, Variable$)
    Aendern Summand2$, "Zeitangabe.Status", SPACE$(1)
    StatusNeu Summand2$
    Addieren Summand1$, "U", "-", Summand2$, 1, 1, "-"
    IF Anders% MOD 2 = 1 THEN
        Summand2$ = Teil$(Summand1$, "Zeitangabe.Minimum")
        Aendern Summand1$, "Zeitangabe.Maximum", Summand2$
    
```

```

ELSE
    Summand2$ = Teil$(Summand1$, "Zeitangabe.Maximum")
    Aendern Summand1$, "Zeitangabe.Minimum", Summand2$
END IF
StatusNeu (Summand1$)
Differenz$ = Summand1$
Laenge% = SIZEOF$( "Zeitangabe " )
IF Differenz$ <> SPACE$(Laenge%) THEN
    Ausgabe 2, "Zeitoffset: " + Differenz$, "", 0, "M", 1, 1, 3
    FOR z& = 1 TO GEL&
        Text$ = " Eintrag " + Namenseintrag$(z&) + " ( "
        Text$ = Text$ + LTRIM$(STR$(z&)) + "/" + LTRIM$(STR$(GEL&))
        Text$ = Text$ + ") wird geprüft. "
        Ausgabe 1, "", Text$, 0, "M", 1, 1, 23
        Daten$ = Laden$(Eingabe%, DatenLaenge%, z&)
        FOR y% = 1 TO 6
            Summand1$ = Differenz$
            Variable$ = "Daten.Datum[ " + LTRIM$(STR$(y%)) + "]"
            Summand2$ = Teil$(Daten$, Variable$)
            IF Summand2$ <> SPACE$(Laenge%) THEN
                Status$ = Teil$(Summand2$, "Zeitangabe.Status")
                Aendern Summand2$, "Zeitangabe.Status", SPACE$(1)
                StatusNeu Summand2$
                Addieren Summand1$, ">", "+", Summand2$, 0, 1, " "
                IF LTRIM$(Summand1$) = "" THEN
                    y% = 6
                    z& = GEL&
                ELSE
                    Aendern Summand1$, "Zeitangabe.Status", Status$
                    Aendern Daten$, Variable$, Summand1$
                END IF
            END IF
        NEXT y%
    NEXT z&
    IF LTRIM$(Summand1$) <> "" THEN
        FOR z& = 1 TO GEL&
            Text$ = " Eintrag " + Namenseintrag$(z&) + " ( "
            Text$ = Text$ + LTRIM$(STR$(z&)) + "/" + LTRIM$(STR$(GEL&))
            Text$ = Text$ + ") wird umdatiert. "
            Ausgabe 1, "", Text$, 0, "M", 1, 1, 23
            Daten$ = Laden$(Eingabe%, DatenLaenge%, z&)
            FOR y% = 1 TO 6
                Summand1$ = Differenz$
                Variable$ = "Daten.Datum[ " + LTRIM$(STR$(y%)) + "]"
                Summand2$ = Teil$(Daten$, Variable$)
                IF Summand2$ <> SPACE$(Laenge%) THEN
                    Status$ = Teil$(Summand2$, "Zeitangabe.Status")
                    Aendern Summand2$, "Zeitangabe.Status", SPACE$(1)
                    StatusNeu Summand2$
                    Addieren Summand1$, ">", "+", Summand2$, 0, 1, " "
                    Aendern Summand1$, "Zeitangabe.Status", Status$
                    Aendern Daten$, Variable$, Summand1$
                END IF
            NEXT y%
            Aendern Daten$, "Daten.Ende", " | " + CHR$(13) + CHR$(10)
            Schreiben Eingabe%, DatenLaenge%, z&, Daten$
        NEXT z&
    END IF

```

```

        ELSE
            Ausgabe 2, "Zeitoffset: nicht vorhanden.", "", 0, "M", 1, 1, 3
            Pause
        END IF
    END IF
    CLOSE #Eingabe%
RETURN 'W8 Datei umdatieren _____'

'=====
W9:      'Datei ausdrucken:
'=====
    GOSUB Dateiname
    IF Erfolg% <> 0 THEN

91  Konfiguration% = FREEFILE
    OPEN Arbeitsplatz$ + "DRUCKEN.CFG" FOR OUTPUT AS #Konfiguration%
    PRINT #Konfiguration%, EingabeDatei$
    PRINT #Konfiguration%, STR$(GEL&)
    PRINT #Konfiguration%, "HISTORIK.BAS "
    CLOSE #Konfiguration%
92  CHAIN "DRUCKEN.BAS "
    END IF
RETURN 'W9 Datei ausdrucken _____'

'=====
Dateiname:
'=====
    ' Fragt den Dateinamen ab.
    ' Die Übersicht über die Dateinamen muß unabhängig vorliegen!
    ' Die Übersicht über die Dateinamen ist über die Pfeiltasten wählbar.
    ,
    ' Bearbeitung:
    ' 8. 1.2003 - 8. 2.2003: Norbert Südland
    ' 14.11.2006:      Norbert Südland
    ' -----

GOSUB VerzeichnisUebersicht
Erfolg% = -1
WHILE Erfolg% < 0
    LOCATE 22, 1
    PRINT SPACE$(80);
    LOCATE 23, 1
    PRINT SPACE$(80);

    Eingabe% = FREEFILE
    OPEN Arbeitsplatz$ + "HQL.DAT" FOR INPUT AS #Eingabe%
    LINE INPUT #Eingabe%, Text$
    z& = 0
    IF Vorgabe& = 0 THEN
        WHILE EOF(Eingabe%) = 0
            LINE INPUT #Eingabe%, Zeile$
            z& = z& + 1
        WEND
        Vorgabe& = z&
    ELSE
        FOR z& = 1 TO Vorgabe&
100      LINE INPUT #Eingabe%, Zeile$

```

```

        IF z& = 1 THEN
            Puffer$ = Zeile$
        END IF
    NEXT z&
END IF
Zeile$ = LEFT$(Zeile$, 8)
CLOSE #Eingabe%

LOCATE 23, 1
PRINT SPACE$(80);
IF LEN(Text$) > 80 - 19 THEN
    Text$ = LEFT$(Text$, 80 - 19 - 20) + " ... " + RIGHT$(Text$, 13)
END IF
Ausgabe 1, "Daten-Verzeichnis: ", Text$, 0, "M", 1, 1, 22
Ausgabe 2, "Dateiname: ", Zeile$ + ".HQL", 0, "M", 1, 1, 23
Puffer$ = "[ " + CHR$(25) + " ], [ " + CHR$(24)
Puffer$ = Puffer$ + " ], [ " + CHR$(17) + "␣" ], [ Einfg ], [ ESC ]"
Ausgabe 1, "Eingabe-Auswahl: ", Puffer$, 0, "M", 1, 1, 25

IF Zeile$ = SPACE$(8) THEN
    Antwort$ = CHR$(0) + CHR$(82)
ELSE
    Antwort$ = Taste$
END IF
SELECT CASE ASC(Antwort$)
CASE 10, 13
    EingabeDatei$ = Zeile$
    Erfolg% = 1
CASE 27
    Erfolg% = 0
END SELECT
IF LEN(Antwort$) = 2 THEN
    Antwort$ = RIGHT$(Antwort$, 1)
    SELECT CASE ASC(Antwort$)
    CASE 71
        'Home
        Vorgabe& = 1
    CASE 72, 75
        'Curser hoch, Curser links
        Vorgabe& = Vorgabe& - 1
    CASE 77, 80
        'Curser rechts, Curser runter
        Vorgabe& = Vorgabe& + 1
    CASE 79
        'End
        Vorgabe& = 0
    CASE 82
        'Einfg
        LOCATE 25, 1
        PRINT SPACE$(80);
        LOCATE 23, 35
        Laenge% = STRLEN%(Zeile$, ".") + 1
        IF Laenge% = 1 THEN
            Zeile$ = SPACE$(8)
            Laenge% = 9
        END IF
        Zeile$ = Editor$(LEFT$(Zeile$, Laenge% - 1), 8, 2, Erfolg%)
        IF Erfolg% = 0 THEN
            Erfolg% = -1
        END IF
    END SELECT
END IF
WEND

```

```

EingabeDatei$ = UCASE$(Zeile$) + ".HQL"
LOCATE 23, 35
PRINT SPACE$(40)
IF Erfolg% = 0 THEN
    GOTO DateinameEnde
END IF
LOCATE 23, 18
PRINT "Datendatei "; EingabeDatei$; " ";

GOSUB EingabeDateiAuf
IF HistAuswahl% = 2 AND GEL& >= 16777215 THEN
    PRINT "ist vollständig belegt."
    Pause
ELSE
    PRINT "hat "; GEL&; "Dateieinträge."
END IF
IF GEL& = 0 THEN Pause

GOSUB KonfigurationAbspeichern
'=====
DateinameEnde:
'=====
    KILL Arbeitsplatz$ + "HQL.DAT"
RETURN 'Dateiname _____'

'=====
VerzeichnisUebersicht:
'=====
' Erstellt eine einheitliche Liste des Verzeichnisinhalts.
'
' Bearbeitung:
' 8. 1.2003 - 8. 2.2003: Norbert Südland
' 14.11.2006:           Norbert Südland
' 11. 8.2009:           Norbert Südland: Anpassung an die DOS-Box
'-----
'

CLOSE #Eingabe%
Erfolg% = 0
z& = 0
Vorgabe& = 0
WHILE Vorgabe& = 0 AND Erfolg% < 2

    'Arbeitet bei verschiedenen DOS-Versionen unterschiedlich:
    '-----
    SELECT CASE Erfolg%
    CASE 0
        SHELL "DIR " + Arbeitsplatz$ + ".* > " + Arbeitsplatz$ + "DIR.DAT"
    CASE 1
        SHELL "DIR *.* > " + Arbeitsplatz$ + "DIR.DAT"
    END SELECT

    'Liste mit Dateinamen ".*.HQL" erstellen:
    '-----
    DirDat% = FREEFILE
    OPEN Arbeitsplatz$ + "DIR.DAT" FOR BINARY ACCESS READ AS #DirDat%
    HQLDat% = FREEFILE
    OPEN Arbeitsplatz$ + "HQL.DAT" FOR OUTPUT AS #HQLDat%

```

```

Ausgabe 1, "Kopierversuch der *.HQL-Dateien...", "", 0, "M", 1, 1, 22
WHILE EOF(DirDat%) = 0
    LINEINPUT DirDat%, Zeile$
    IF INSTR(2, Zeile$, ":\") THEN
        Gefunden% = INSTR(2, Zeile$, ":\") - 2
        Puffer$ = RIGHT$(Zeile$, LEN(Zeile$) - Gefunden%)
        IF RIGHT$(Puffer$, 2) = "\" THEN
            Puffer$ = LEFT$(Puffer$, LEN(Puffer$) - 2)
        END IF
        PRINT #HQLDat%, Puffer$
    ELSEIF MID$(Zeile$, 9, 4) = " HQL" THEN
        Zeile$ = LEFT$(Zeile$, 8) + ".HQL"
        z& = z& + 1
        PRINT #HQLDat%, Zeile$
        IF Zeile$ = EingabeDatei$ THEN
            Vorgabe& = z&
        END IF
        IF Erfolg% = 1 THEN

            'Ungeprüfte Kopie erstellen:
            '-----'
            Eingabe% = FREEFILE
            OPEN Zeile$ FOR BINARY ACCESS READ AS #Eingabe%
            Kopie% = FREEFILE
            OPEN Arbeitsplatz$ + Zeile$ FOR BINARY ACCESS WRITE AS #Kopie%
            Daten$ = SPACE$(SIZEOF("Daten"))
            z& = 0
            WHILE EOF(Eingabe%) = 0
117      GET #Eingabe%, z& * DatenLaenge% + 1, Daten$
            Puffer$ = Teil$(Daten$, "Daten.Name")
            IF ASC(LEFT$(LTRIM$(Puffer$), 1)) <> 0 THEN

                'Alte Dateien auf aktuelles Datei-Format anpassen:
                '-----'
                Puffer$ = Teil$(Daten$, "Daten.V[1].ZP1 ")
                Aendern Daten$, "Daten.V[1].ZP1 ", FuerDatei$(Puffer$)
                Puffer$ = Teil$(Daten$, "Daten.V[1].Direkt ")
                Aendern Daten$, "Daten.V[1].Direkt ", FuerDatei$(Puffer$)
                Puffer$ = Teil$(Daten$, "Daten.V[1].ZP2 ")
                Aendern Daten$, "Daten.V[1].ZP2 ", FuerDatei$(Puffer$)

                Puffer$ = Teil$(Daten$, "Daten.V[2].ZP1 ")
                Aendern Daten$, "Daten.V[2].ZP1 ", FuerDatei$(Puffer$)
                Puffer$ = Teil$(Daten$, "Daten.V[2].Direkt ")
                Aendern Daten$, "Daten.V[2].Direkt ", FuerDatei$(Puffer$)
                Puffer$ = Teil$(Daten$, "Daten.V[2].ZP2 ")
                Aendern Daten$, "Daten.V[2].ZP2 ", FuerDatei$(Puffer$)

                Puffer$ = Teil$(Daten$, "Daten.B[1].ZP1 ")
                Aendern Daten$, "Daten.B[1].ZP1 ", FuerDatei$(Puffer$)
                Puffer$ = Teil$(Daten$, "Daten.B[1].ZP2 ")
                Aendern Daten$, "Daten.B[1].ZP2 ", FuerDatei$(Puffer$)

                Puffer$ = Teil$(Daten$, "Daten.B[2].ZP1 ")
                Aendern Daten$, "Daten.B[2].ZP1 ", FuerDatei$(Puffer$)
                Puffer$ = Teil$(Daten$, "Daten.B[2].ZP2 ")
                Aendern Daten$, "Daten.B[2].ZP2 ", FuerDatei$(Puffer$)

```

```

        Puffer$ = Teil$(Daten$, "Daten.Gleichzeitig")
        Puffer$ = FuerDatei$(Puffer$)
        Aendern Daten$, "Daten.Gleichzeitig", Puffer$

        Aendern Daten$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)

        PUT #Kopie%, z& * DatenLaenge% + 1, Daten$
    END IF
    z& = z& + 1
WEND
    CLOSE #Kopie%
    CLOSE #Eingabe%
END IF
END IF
WEND
    CLOSE #HQLDat%
    CLOSE #DirDat%
    KILL Arbeitsplatz$ + "DIR.DAT"
    Erfolg% = Erfolg% + 1
WEND

RETURN 'VerzeichnisUebersicht _____'

'=====
EingabeDateiAuf:
'=====
' Öffnet die Eingabe-Datei.
' Nur wenn die Eingabe-Datei im Arbeitsverzeichnis leer ist, wird
' im aufrufenden Verzeichnis nach einer Kopier-Möglichkeit gesucht.
'
' Bearbeitung: 15.12.2002 - 20. 1.2003 Norbert Südland
'-----
Laenge% = DatenLaenge%
Daten$ = SPACE$(DatenLaenge%)
KopieDatei$ = Arbeitsplatz$ + EingabeDatei$      'Arbeitsplatz$ nutzen!
GEL& = 0
Eingabe% = FREEFILE
101 OPEN KopieDatei$ FOR BINARY ACCESS READ WRITE AS #Eingabe%
    Ausstieg% = 0
    Daten$ = SPACE$(DatenLaenge%)
    WHILE Ausstieg% = 0
102     GET #Eingabe%, GEL& * DatenLaenge% + 1, Daten$
        IF GEL& = 0 THEN GOSUB DateiDokumentation
        IF ASC(LEFT$(LTRIM$(Teil$(Daten$, "Daten.Name")), 1)) <> 0 THEN
            Aendern Daten$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
            PUT #Eingabe%, GEL& * DatenLaenge% + 1, Daten$ 'Schreibtest!
            GEL& = GEL& + 1
        ELSE
            Ausstieg% = 1
        END IF
    WEND
103     GEL& = GEL& - 1
        IF GEL& > 0 THEN RETURN 'Die Datei ist vorhanden und beschreibbar!
104 CLOSE #Eingabe%

'Falls die Datei nicht beschreibbar ist, wird eine Kopie erstellt:
'-----
GEL& = 0

```

```

    Eingabe% = FREEFILE
105 OPEN EingabeDatei$ FOR BINARY ACCESS READ AS #Eingabe%
    Kopie% = FREEFILE
106 OPEN KopieDatei$ FOR BINARY ACCESS WRITE AS #Kopie%
    Ausstieg% = 0
    Daten$ = SPACE$(DatenLaenge%)
    WHILE EOF(Eingabe%) = 0 AND Ausstieg% = 0
107     GET #Eingabe%, GEL& * DatenLaenge% + 1, Daten$
        IF GEL& = 0 THEN GOSUB DateiDokumentation
        IF ASC(LEFT$(LTRIM$(Teil$(Daten$, "Daten.Name")), 1)) <> 0 THEN

            'Alte Dateien auf aktuelles Datei-Format anpassen:
            '-----'
            Puffer$ = Teil$(Daten$, "Daten.V[1].ZP1 ")
            Aendern Daten$, "Daten.V[1].ZP1 ", FuerDatei$(Puffer$)
            Puffer$ = Teil$(Daten$, "Daten.V[1].Direkt ")
            Aendern Daten$, "Daten.V[1].Direkt ", FuerDatei$(Puffer$)
            Puffer$ = Teil$(Daten$, "Daten.V[1].ZP2 ")
            Aendern Daten$, "Daten.V[1].ZP2 ", FuerDatei$(Puffer$)

            Puffer$ = Teil$(Daten$, "Daten.V[2].ZP1 ")
            Aendern Daten$, "Daten.V[2].ZP1 ", FuerDatei$(Puffer$)
            Puffer$ = Teil$(Daten$, "Daten.V[2].Direkt ")
            Aendern Daten$, "Daten.V[2].Direkt ", FuerDatei$(Puffer$)
            Puffer$ = Teil$(Daten$, "Daten.V[2].ZP2 ")
            Aendern Daten$, "Daten.V[2].ZP2 ", FuerDatei$(Puffer$)

            Puffer$ = Teil$(Daten$, "Daten.B[1].ZP1 ")
            Aendern Daten$, "Daten.B[1].ZP1 ", FuerDatei$(Puffer$)
            Puffer$ = Teil$(Daten$, "Daten.B[1].ZP2 ")
            Aendern Daten$, "Daten.B[1].ZP2 ", FuerDatei$(Puffer$)

            Puffer$ = Teil$(Daten$, "Daten.B[2].ZP1 ")
            Aendern Daten$, "Daten.B[2].ZP1 ", FuerDatei$(Puffer$)
            Puffer$ = Teil$(Daten$, "Daten.B[2].ZP2 ")
            Aendern Daten$, "Daten.B[2].ZP2 ", FuerDatei$(Puffer$)

            Puffer$ = Teil$(Daten$, "Daten.Gleichzeitig")
            Aendern Daten$, "Daten.Gleichzeitig", FuerDatei$(Puffer$)

            Aendern Daten$, "Daten.Ende", " " + CHR$(13) + CHR$(10)
108     PUT #Kopie%, GEL& * DatenLaenge% + 1, Daten$ 'Schreibtest!
        GEL& = GEL& + 1
    ELSE
109     Ausstieg% = 1
    END IF
    WEND
    GEL& = GEL& - 1
    CLOSE #Kopie%
    CLOSE #Eingabe%
    OPEN KopieDatei$ FOR BINARY ACCESS READ WRITE AS #Eingabe%
    RETURN 'EingabeDateiAuf _____'

'=====
Loeschen:
'=====
' Löscht einen Datei-Eintrag:
'

```

```

' Bearbeitung:
' 21. 1.2003: Norbert Südland
' Überprüfung:
' 15.12.2002 - 21. 1.2003:      Norbert Südland
' -----
IF Auswahl& <> 0 THEN
  Text$ = "Eintrag " + Namenseintrag$(Auswahl&) + "wird gelöscht."
  Ausgabe 1, Text$, "", 0, "M", 1, 1, 24

  'Zwischenkopie erstellen:
  '-----'
  KopieDatei$ = LEFT$(EingabeDatei$, STRLEN$(EingabeDatei$, ".") + ".KOP"
  KopieDatei$ = Arbeitsplatz$ + KopieDatei$
  Kopie% = FREEFILE
  Laenge% = DatenLaenge%
  OPEN KopieDatei$ FOR BINARY ACCESS READ WRITE AS #Kopie%
  Daten$ = SPACE$(DatenLaenge%)
  GOSUB DateiDokumentation
  PUT #Kopie%, 1, Daten$
  FOR z& = 1 TO GEL&
    Daten$ = Laden$(Eingabe%, DatenLaenge%, z&)
    Aendern Daten$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
    Schreiben Kopie%, DatenLaenge%, z&, Daten$
  NEXT z&
  CLOSE #Eingabe%

  'Original löschen:
  '-----'
  KILL Arbeitsplatz$ + EingabeDatei$

  'Neue Datei mit weniger Einträgen erzeugen:
  '-----'
  KopieDatei$ = Arbeitsplatz$ + EingabeDatei$      'Arbeitsplatz$ nutzen!
  OPEN KopieDatei$ FOR BINARY ACCESS READ WRITE AS #Eingabe%
  Daten$ = SPACE$(DatenLaenge%)
  GOSUB DateiDokumentation
  PUT #Eingabe%, 1, Daten$
  Offset% = 0
  FOR z& = 1 TO GEL&
    IF z& = Auswahl& THEN
      Offset% = 1
    ELSE
      Daten$ = Laden$(Kopie%, DatenLaenge%, z&)
      Aendern Daten$, "Daten.Ende", "|" + CHR$(13) + CHR$(10)
      Schreiben Eingabe%, DatenLaenge%, z& - Offset%, Daten$
    END IF
  NEXT z&
  GEL& = GEL& - Offset%
  CLOSE #Kopie%

  'Zwischendatei löschen:
  '-----'
  KopieDatei$ = LEFT$(EingabeDatei$, STRLEN$(EingabeDatei$, ".") + ".KOP"
  KopieDatei$ = Arbeitsplatz$ + KopieDatei$
  KILL KopieDatei$
END IF
RETURN 'Löschen _____'

```

```

'=====
DateiDokumentation:
'=====
' Erstellt für den nullten Datei-Eintrag eine Dokumentationszeile.
'
' Bearbeitung:
' 21. 1.2003: Norbert Südland
'-----
FOR x% = 1 TO 6
  Puffer$ = "Minimum[ " + LTRIM$(STR$(x%)) + "]" "
  Aendern Daten$, "Daten.Datum[ " + LTRIM$(STR$(x%)) + "].Minimum", Puffer$
  Puffer$ = "Maximum[ " + LTRIM$(STR$(x%)) + "]" "
  Aendern Daten$, "Daten.Datum[ " + LTRIM$(STR$(x%)) + "].Maximum", Puffer$
NEXT x%
Aendern Daten$, "Daten.Name", "Name /Ereignis"
Aendern Daten$, "Daten.Quelle", "1.Quellenangabe"
FOR x% = 1 TO 2
  Puffer$ = LTRIM$(STR$(x%)) + ".Vorgänger "
  Aendern Daten$, "Daten.V[ " + LTRIM$(STR$(x%)) + "]", Puffer$
  Puffer$ = LTRIM$(STR$(x%)) + ".Datum"
  Aendern Daten$, "Daten.B[ " + LTRIM$(STR$(x%)) + "].Datum", Puffer$
  Puffer$ = LTRIM$(STR$(x%)) + ".Bezugsname"
  Aendern Daten$, "Daten.B[ " + LTRIM$(STR$(x%)) + "].Name", Puffer$
  Puffer$ = LTRIM$(STR$(x%)) + ".Dauer"
  Aendern Daten$, "Daten.B[ " + LTRIM$(STR$(x%)) + "].Dauer", Puffer$
  Puffer$ = LTRIM$(STR$(x% + 1)) + ".Quellenangabe"
  Aendern Daten$, "Daten.B[ " + LTRIM$(STR$(x%)) + "].Quelle", Puffer$
NEXT x%
Aendern Daten$, "Daten.Ende", " " + CHR$(13) + CHR$(10)
RETURN 'DateiDokumentation _____

```

SUBROUTINEN UND FUNKTIONEN

```

'=====
FUNCTION Abfrage$(Text$, Vorgabe$, Eingabeerfolg%)
'=====

```

```

' Gibt eine Frage auf dem Bildschirm aus und wartet auf Ja oder Nein.
'
' Bearbeitung:
' 10. 9.2001: Norbert Südland, München
' 18. 3.2017: Norbert Südland, Aalen
'-----

```

```

DIM Antwort$

```

```

DIM x%

```

```

DIM y%

```

```

IF Vorgabe$ <> "N" AND Vorgabe$ <> "J" THEN Vorgabe$ = "N"

```

```

LOCATE 25, 1: PRINT SPACE$(79);

```

```

Ausgabe 2, Text$ + " [J/N]", Vorgabe$, -1, "M", 1, 1, 25

```

```

x% = CSRLIN

```

```

y% = POS(0) - 1

```

```

DO

```

```

  LOCATE x%, y%

```

```

  Antwort$ = Editor$(Vorgabe$, 1, 2, Eingabeerfolg%)

```

```

    IF Eingabeerfolg% = 1 THEN
        IF Antwort$ = "j" THEN Antwort$ = "J"
        IF Antwort$ = "n" THEN Antwort$ = "N"
    END IF
    LOOP UNTIL Antwort$ = "J" OR Antwort$ = "N" OR Eingabeerfolg% = 0
    PRINT Antwort$;
    Abfrage$ = Antwort$
END FUNCTION 'Abfrage$ _____'

'=====
SUB Addieren (Summand1$, Art$, Richtung$, Vorgabe$, OZahl1%, OZahl2%, Weiter$)
'=====
'
'  Bearbeitung:
'  27. 3.2003: Norbert Südland
'-----
DIM Summand2$
DIM Status1$
DIM Status2$
DIM Status$
DIM Termin1$
DIM Termin2$
DIM Groesse%
DIM v0%, v1%, v2%, v3%, v4%, v5%, v6%
DIM j0%, j1%, j2%, j3%, j4%, j5%, j6%
DIM m0%, m1%, m2%, m3%, m4%, m5%, m6%
DIM t0%, t1%, t2%, t3%, t4%, t5%, t6%
DIM Start%
DIM Ende%
DIM z%

'Vorbereitung:
'-----
Summand2$ = Vorgabe$
StatusNeu Summand1$
Status1$ = Teil$(Summand1$, "Zeitangabe.Status")
StatusNeu Summand2$
Status2$ = Teil$(Summand2$, "Zeitangabe.Status")
Groesse% = SIZEOF$("Zeitangabe")
IF Summand1$ = SPACE$(Groesse%) AND Summand2$ = SPACE$(Groesse%) THEN
    GOTO AddierenEnde
END IF

'Art der Addition beachten:
'-----
IF Art$ = "U" THEN      'Vereinigungsmenge
    Status$ = Status1$
    IF Status$ <> Status2$ THEN
        Status$ = " "
    END IF
ELSE
    IF Art$ = ">" THEN 'Summand2$ gibt alles vor
        Status$ = Status2$
        IF Summand2$ = SPACE$(Groesse%) THEN
            Summand1$ = Summand2$
            GOTO AddierenEnde
        END IF

```

```

ELSE
    GOTO AddierenEnde
END IF
END IF

'eigentliche Addition:
'-----'
TerminZuZahl Teil$(Summand1$, "Zeitangabe.Minimum"), v3%, j3%, m3%, t3%
TerminZuZahl Teil$(Summand2$, "Zeitangabe.Minimum"), v4%, j4%, m4%, t4%
TerminZuZahl Teil$(Summand1$, "Zeitangabe.Maximum"), v5%, j5%, m5%, t5%
TerminZuZahl Teil$(Summand2$, "Zeitangabe.Maximum"), v6%, j6%, m6%, t6%
IF OZahl1% <> 0 THEN
    ZeitSchieben v3%, j3%, m3%, t3%, Status1$, v5%, j5%, m5%, t5%, "-"
END IF
IF OZahl2% <> 0 THEN
    ZeitSchieben v4%, j4%, m4%, t4%, Status2$, v6%, j6%, m6%, t6%, "-"
END IF
Umkehren v4%, j4%, m4%, t4%, v6%, j6%, m6%, t6%, Richtung$
SELECT CASE Status$
CASE ">"
    Start% = 1
    Ende% = 1
CASE "<"
    Start% = 2
    Ende% = 2
CASE " "
    Start% = 1
    Ende% = 2
END SELECT
FOR z% = Start% TO Ende%
    IF z% = 1 THEN
        v1% = v3%: j1% = j3%: m1% = m3%: t1% = t3%
        v2% = v4%: j2% = j4%: m2% = m4%: t2% = t4%
    ELSE
        v1% = v5%: j1% = j5%: m1% = m5%: t1% = t5%
        v2% = v6%: j2% = j6%: m2% = m6%: t2% = t6%
    END IF
    t0% = t1% + t2%
    m0% = m1% + m2%
    j0% = v1% * j1% + v2% * j2%
    v0% = 1
    Restklasse v0%, j0%, m0%, t0%
    IF ABS(j0%) >= 10000 THEN
        z% = Ende%
    ELSE
        IF z% = 1 THEN
            v3% = v0%: j3% = j0%: m3% = m0%: t3% = t0%
        ELSE
            v5% = v0%: j5% = j0%: m5% = m0%: t5% = t0%
        END IF
    END IF
NEXT z%
IF ABS(j0%) >= 10000 THEN GOTO AddierenEnde
IF (OZahl1% <> 0 OR OZahl2% <> 0) AND Weiter$ <> "-" THEN
    ZeitSchieben v3%, j3%, m3%, t3%, Status1$, v5%, j5%, m5%, t5%, "+"
END IF
SELECT CASE Status$
CASE ">"

```

```

Termin1$ = ZahlZuTermin$(v3%, j3%, m3%, t3%)
Termin2$ = SPACE$(SIZEOF%("Termin"))
CASE "<"
Termin1$ = SPACE$(SIZEOF%("Termin"))
Termin2$ = ZahlZuTermin$(v5%, j5%, m5%, t5%)
CASE " "
Termin1$ = ZahlZuTermin$(v3%, j3%, m3%, t3%)
Termin2$ = ZahlZuTermin$(v5%, j5%, m5%, t5%)
END SELECT
Aendern Summand1$, "Zeitangabe.Minimum", Termin1$
Aendern Summand1$, "Zeitangabe.Status", Status$
Aendern Summand1$, "Zeitangabe.Maximum", Termin2$

'=====
AddierenEnde:
'=====
IF ABS(j0%) >= 10000 THEN
Summand1$ = SPACE$(SIZEOF%("Zeitangabe"))
END IF
END SUB 'Addieren _____

'=====
SUB Aendern (Daten$, Variable$, Inhalt$)
'=====
' Fügt 'Inhalt$' an der richtigen Stelle in 'Daten$' ein.
'
' Bearbeitung:
' 4. 9.2001 - 21. 1.2003:      Norbert Südland
'-----

DIM Suche$
DIM Laenge%
DIM p%
DIM Anfang%
DIM Ende%
DIM Wo%
DIM np%
DIM i%

Laenge% = LEN(Variable$)
p% = STRLEN%(Variable$, ".")
IF p% = Laenge% THEN
Suche$ = Variable$
ELSE
p% = p% + 1
p% = p% + STRLEN%(MID$(Variable$, p% + 1, Laenge% - p%), ".")
IF p% < Laenge% THEN
Suche$ = LEFT$(Variable$, p%)
ELSE
Suche$ = Variable$
END IF
END IF
Anfang% = 1
DO
Ende% = GlobVarAnzahl%
Wo% = Nummer%(Suche$, 1, Ende%)
IF Wo% = 0 THEN
Pause      'Suche$ nicht gefunden!

```

```

        ERROR 111
    END IF
    Anfang% = Anfang% + GVAnfang%(Wo%) - 1
    IF p% < Laenge% THEN
        p% = p% + 1
        np% = STRLEN%(MID$(Variable$, p% + 1, Laenge% - p%), ".")
        Suche$ = GVTyp$(Wo%) + MID$(Variable$, p%, np% + 1)
        p% = p% + np%
    ELSE
        p% = Laenge% + 1
    END IF
    LOOP WHILE p% <= Laenge%

    i% = LEN(Inhalt$)
    IF i% < GVLaenge%(Wo%) THEN
        Inhalt$ = SPACE$(GVLaenge%(Wo%) - i%) + Inhalt$
    ELSE
        IF i% > GVLaenge%(Wo%) THEN
            Inhalt$ = LEFT$(Inhalt$, GVLaenge%(Wo%))
        END IF
    END IF

    MID$(Daten$, Anfang%, GVLaenge%(Wo%)) = Inhalt$
END SUB 'Aendern _____'

'=====
FUNCTION AktuellesDatum$
'=====
' Erstellt die Ausgabezeile mit dem aktuellen Datum.
'
' Bearbeitung:
' 4. 9.2001: Norbert Südland, München
'-----
DIM Text$

SELECT CASE GlobWochentag%
CASE 1
    Text$ = "Sonntag"
CASE 2
    Text$ = "Montag"
CASE 3
    Text$ = "Dienstag"
CASE 4
    Text$ = "Mittwoch"
CASE 5
    Text$ = "Donnerstag"
CASE 6
    Text$ = "Freitag"
CASE 7
    Text$ = "Samstag"
END SELECT
Text$ = Text$ + " (" + LTRIM$(STR$(GlobWochentag%)) + ".Wochentag), den "
Text$ = Text$ + GlobDatum$

AktuellesDatum$ = Text$
END FUNCTION 'AktuellesDatum$ _____'

```

```

'=====
FUNCTION Anlegen% (Daten$, Position&, HistAuswahl%, Eingabeerfolg%)
'=====
'
' Bearbeitung:
' 5. 8.2001 - 4. 9.2001: Norbert Südland, München
'-----
DIM Text$
DIM Antwort$
DIM z%

CLS
Text$ = LTRIM$(STR$(Position&)) + ".Datei-Element:"
Ausgabe 2, Text$, "", 0, "M", 1, 1, 1
IF HistAuswahl% = 2 THEN
    Ausgabe 1, "", "(2)...Datei anlegen /erweitern:", 0, "M", 1, 1, 2
ELSEIF HistAuswahl% = 5 THEN
    Ausgabe 1, "", "(5)...Datei-Eintrag korrigieren:", 0, "M", 1, 1, 2
END IF
FOR z% = 1 TO 36
    ZeilenAusgabe Daten$, z%, 1
NEXT z%
DO
    FOR z% = 1 TO 35
        z% = EinzelEingabe%(Daten$, z%, 2, Eingabeerfolg%)
        IF Eingabeerfolg% = 0 THEN z% = 35
    NEXT z%
    IF Eingabeerfolg% = 1 THEN
        IF z% = 39 THEN z% = EinzelEingabe%(Daten$, 34, 2, Eingabeerfolg%)
        IF Eingabeerfolg% = 1 AND z% < 40 THEN
            Antwort$ = Abfrage$("Daten korrigieren?", "N", Eingabeerfolg%)
        END IF
    END IF
LOOP UNTIL z% = 40 OR Antwort$ = "N" OR Eingabeerfolg% = 0
Anlegen% = z% MOD 40
END FUNCTION 'Anlegen% _____

'=====
SUB Ausgabe (Farbe%, Frage$, Puffer$, Offset%, Art$, Bereich%, Bereiche%, y%)
'=====
' Gibt `Frage$` gefolgt von `Puffer$` in Zeile `y%` aus.
'
' Bedeutung der weiteren übergebenen Parameter:
' `Farbe%` Wahlmodus 1 (normal) oder 2 (hervorgehoben)
' `Offset%` Verschiebungsmöglichkeit innerhalb einer Spalte
' `Bereich%` Gewünschte-Spalte
' `Bereiche%` Spaltenzahl insgesamt
' `Art$` "L" (linksbündig), "M" (mittenzentriert), "R" (rechtsbündig)
'
' Bearbeitung:
' 7. 8.2001 - 4. 9.2001: Norbert Südland und Eckhard Walter, Adelshofen
'-----
DIM Laenge% 'AS INTEGER
DIM x%

SELECT CASE Farbe%

```

```

CASE 1
    COLOR 7, 0
CASE 2
    COLOR 15, 0
END SELECT
Laenge% = LEN(Frage$) + LEN(Puffer$) + 2
IF LEN(Puffer$) = 0 THEN
    Laenge% = Laenge% - 2
END IF
IF Laenge% > 80 / Bereiche% THEN
    Laenge% = 80 / Bereiche%
    IF LEN(Frage$) >= Laenge% THEN        'Programmier-Ungereimtheit!
        Frage$ = LEFT$(Frage$, Laenge%)
        Puffer$ = ""
    ELSE
        Pause
        Puffer$ = LEFT$(Puffer$, LEN(Frage$) - Laenge%)
    END IF
END IF

SELECT CASE Art$
CASE "L"
    x% = INT(Offset% + (Bereich% - 1) * 80 / Bereiche%)
CASE "M"
    x% = INT(Offset% + (Bereich% - .5) * 80 / Bereiche% - Laenge% / 2) + 1
CASE "R"
    x% = INT(Offset% + Bereich% * 80 / Bereiche% - Laenge%)
END SELECT
LOCATE y%, x%
PRINT Frage$;

IF Puffer$ <> "" THEN
    COLOR 0, 7
    SELECT CASE Farbe%
    CASE 1
        PRINT " "; Puffer$; " ";
    CASE 2
        IF Puffer$ <> "" THEN
            x% = POS(0)
            LOCATE y%, x% + 1
            PRINT Puffer$;
        END IF
    END SELECT
END IF

COLOR 7, 0
END SUB 'Ausgabe _____'

'=====
FUNCTION Datieren% (HistAuswahl%, Auswahl&, Termin$)
'=====
'
'  Bearbeitung:
'  5. 8.2001 -  4. 9.2001:    Norbert Südland, München
'  16.12.2002 - 27. 3.2003:    Norbert Südland, Aalen
'-----
DIM Text$      'AS STRING
DIM Daten$

```

```
DIM a$
DIM Variable$
DIM Datum$
DIM j$
DIM m$
DIM t$
DIM Minimum$
DIM Maximum$
DIM Status$
DIM Quelle$
DIM Puffer$
DIM Ende%      'AS INTEGER
DIM x%
DIM y%
DIM z%
DIM Aktuell%
DIM Antwort%
DIM Weiter%
DIM Eingabeerfolg%
DIM Wiederholung%
DIM Laenge%
DIM Quelledazu%

CLS
Termin$ = ""
Text$ = LTRIM$(STR$(Auswahl&)) + ".Dateielement "
Text$ = Text$ + Namenseintrag$(Auswahl&)
SELECT CASE HistAuswahl%
CASE 4, 5:
    Text$ = Text$ + " vordatieren:"
    Ende% = 13
CASE 8:
    Text$ = Text$ + " umdatieren:"
    Ende% = 12
END SELECT
Ausgabe 1, "", Text$, 0, "M", 1, 1, 1
Text$ = "Zum Verlassen des Menüs [ ESC ] drücken."
Ausgabe 2, Text$, "", 0, "M", 1, 1, 25
Daten$ = Laden$(Eingabe%, DatenLaenge%, Auswahl&)
FOR z% = 1 TO Ende%
    Datumsanzeige Daten$, z%, 1, HistAuswahl%
NEXT z%
Aktuell% = 1
Antwort% = -1
DO
    DO
        LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
        Text$ = "Bitte wählen Sie mit den Pfeiltasten [ " + CHR$(24) + " ] oder "
        Text$ = Text$ + " [ " + CHR$(25) + " ] eine Zeile aus "
        Ausgabe 2, Text$, "", 0, "M", 1, 1, 2
        Text$ = "und bestätigen Sie Ihre Wahl mit der Eingabetaste [ENTER]: "
        Ausgabe 2, Text$, "", 0, "M", 1, 1, 3
        Datumsanzeige Daten$, Aktuell%, 2, HistAuswahl%
        a$ = Taste$
        Weiter% = Aktuell%
        SELECT CASE ASC(a$)
        CASE 10, 13
```

```

IF HistAuswahl% = 8 THEN
    Variable$ = "Daten.Datum[" + LTRIM$(STR$(INT((Aktuell% + 1) / 2)))
    IF Aktuell% MOD 2 = 1 THEN
        Variable$ = Variable$ + "].Minimum"
    ELSE
        Variable$ = Variable$ + "].Maximum"
    END IF
    IF Teil$(Daten$, Variable$) = SPACE$(SIZEOF$( "Termin" )) THEN
        a$ = CHR$(10)
    ELSE
        a$ = CHR$(13)
        Antwort% = Aktuell%
    END IF
ELSE
    a$ = CHR$(13)
    Antwort% = Aktuell%
END IF
CASE 27
    a$ = CHR$(13)
    Antwort% = 0
END SELECT
IF LEN(a$) = 2 THEN
    a$ = RIGHT$(a$, 1)
    SELECT CASE ASC(a$)
    CASE 71
        'Home
        Weiter% = 1
    CASE 72
        IF Aktuell% > 1 THEN Weiter% = Aktuell% - 1 ELSE Weiter% = Ende%
    CASE 75
        Weiter% = INT((Aktuell% - 3) / 2) * 2 + 1
        IF Weiter% < 1 THEN Weiter% = Ende%
    CASE 77
        Weiter% = INT((Aktuell% + 1) / 2) * 2 + 1
        IF Weiter% > Ende% THEN Weiter% = 1
    CASE 79
        Weiter% = Ende%
    CASE 80
        IF Aktuell% < Ende% THEN Weiter% = Aktuell% + 1 ELSE Weiter% = 1
    CASE 83
        'Del
        IF Abfrage$( "Eintrag löschen?", "N", Eingabeerfolg% ) = "J" THEN
            IF Aktuell% = 13 AND Eingabeerfolg% = 1 THEN
                Aendern Daten$, "Daten.Quelle", SPACE$(SIZEOF$( "NAME" ))
            ELSE
                IF Eingabeerfolg% = 1 THEN
                    Aktuell% = INT((Aktuell% + 1) / 2)
                    Variable$ = "Daten.Datum[" + LTRIM$(STR$(Aktuell%)) + "]"
                    Aendern Daten$, Variable$, SPACE$(SIZEOF$( "Zeitangabe" ))
                    Aktuell% = (Aktuell% - 1) * 2 + 1
                    Datumsanzeige Daten$, Aktuell% + 1, 1, HistAuswahl%
                END IF
            END IF
        END IF
        Text$ = "Zum Verlassen des Menüs [ ESC ] drücken."
        Ausgabe 2, Text$, "", 0, "M", 1, 1, 25
    END SELECT
    Datumsanzeige Daten$, Aktuell%, 1, HistAuswahl%
    Aktuell% = Weiter%
END IF

```

```
LOOP UNTIL a$ = CHR$(13)
LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
Datumsanzeige Daten$, Aktuell%, 1, HistAuswahl%
IF Antwort% > 0 AND Antwort% < 13 THEN
    Variable$ = "Daten.Datum[ " + LTRIM$(STR$(INT((Antwort% + 1) / 2)))
    Variable$ = Variable$ + "]"
    IF Antwort% MOD 2 = 1 THEN
        Datum$ = Teil$(Daten$, Variable$ + ".Minimum")
    ELSE
        Datum$ = Teil$(Daten$, Variable$ + ".Maximum")
    END IF
    j$ = Teil$(Datum$, "Termin.VZ") + Teil$(Datum$, "Termin.Jahr")
    m$ = Teil$(Datum$, "Termin.Monat")
    t$ = Teil$(Datum$, "Termin.Tag")
    x% = CSRLIN
    y% = POS(0) - SIZEOF("Termin") - 1

    'Eingabe der Jahreszahl:
    '-----'
    Text$ = "Eingabe der Jahreszahl (bis vierstellig) mit Vorzeichen."
    Ausgabe 2, Text$, "", 0, "M", 1, 1, 2
    IF HistAuswahl% = 8 THEN
        Text$ = "0 ist ungültig."
    ELSE
        Text$ = "0 bedeutet Löschen."
    END IF
    Ausgabe 2, Text$, "", 0, "M", 1, 1, 3
    Wiederholung% = 1
    DO
        LOCATE x%, y%
        j$ = Editor$(j$, 5, 1, Eingabeerfolg%)
        IF Eingabeerfolg% = 1 THEN
            Wiederholung% = 0
            IF VAL(j$) < 0 THEN
                j$ = LTRIM$(STR$(-VAL(j$)))
                j$ = "-" + SPACE$(4 - LEN(j$)) + j$
            END IF
        END IF
        IF VAL(j$) = 0 THEN
            IF HistAuswahl% = 8 THEN
                j$ = Teil$(Datum$, "Termin.VZ") + Teil$(Datum$, "Termin.Jahr")
                Wiederholung% = 1
                Eingabeerfolg% = 1
            ELSE
                j$ = SPACE$(6)
                Wiederholung% = 0
            END IF
        END IF
        IF VAL(j$) < -9999 OR VAL(j$) > 9999 THEN
            j$ = Teil$(Datum$, "Termin.VZ") + Teil$(Datum$, "Termin.Jahr")
            Wiederholung% = 1
            Eingabeerfolg% = 1
        END IF
    LOOP UNTIL Wiederholung% = 0 OR Eingabeerfolg% = 0
    IF VAL(j$) <> 0 THEN j$ = j$ + "."
    COLOR 0, 7: PRINT j$;
    LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
    IF Eingabeerfolg% = 0 THEN GOTO DatierenSchleifeEnde
```

```
IF VAL(j$) = 0 THEN
    m$ = SPACE$(3)
    t$ = SPACE$(3)
    GOTO DatumAendern
END IF

'Eingabe der Monatszahl:
'-----'
Text$ = "Eingabe der Monatszahl (zwischen 1 und 12 einschließlic)."
Ausgabe 2, Text$, "", 0, "M", 1, 1, 2
IF HistAuswahl% = 8 THEN
    Text$ = "0 ist ungültig."
ELSE
    Text$ = "0 bedeutet Löschen."
END IF
Ausgabe 2, Text$, "", 0, "M", 1, 1, 3
Wiederholung% = 1
DO
    LOCATE x%, y% + 6
    m$ = Editor$(m$, 2, 1, Eingabeerfolg%)
    IF Eingabeerfolg% = 1 THEN Wiederholung% = 0
    IF VAL(m$) = 0 THEN
        IF HistAuswahl% = 8 THEN
            m$ = Teil$(Datum$, "Termin.Monat ")
            Wiederholung% = 1
            Eingabeerfolg% = 1
        ELSE
            m$ = SPACE$(3)
        END IF
    END IF
    IF VAL(m$) < 0 OR VAL(m$) > 12 THEN
        m$ = Teil$(Datum$, "Termin.Monat ")
        Wiederholung% = 1
        Eingabeerfolg% = 1
    END IF
LOOP UNTIL Wiederholung% = 0 OR Eingabeerfolg% = 0
IF VAL(m$) <> 0 THEN m$ = m$ + "."
COLOR 0, 7: PRINT m$;
LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
IF Eingabeerfolg% = 0 THEN GOTO DatierenSchleifeEnde
IF VAL(m$) = 0 THEN
    j$ = SPACE$(6)
    t$ = SPACE$(3)
    GOTO DatumAendern
END IF

'Eingabe der Tageszahl:
'-----'
Text$ = "Eingabe der Tageszahl (zwischen 1 und 30 einschließlic)."
Ausgabe 2, Text$, "", 0, "M", 1, 1, 2
IF HistAuswahl% = 8 THEN
    Text$ = "0 ist ungültig."
ELSE
    Text$ = "0 bedeutet Löschen."
END IF
Ausgabe 2, Text$, "", 0, "M", 1, 1, 3
Wiederholung% = 1
DO
```

```

LOCATE x%, y% + 9
t$ = Editor$(t$, 2, 1, Eingabeerfolg%)
IF Eingabeerfolg% = 1 THEN Wiederholung% = 0
IF VAL(t$) = 0 THEN
    IF HistAuswahl% = 8 THEN
        t$ = Teil$(Datum$, "Termin.Tag")
        Wiederholung% = 1
        Eingabeerfolg% = 1
    ELSE
        t$ = SPACE$(3)
    END IF
END IF
IF VAL(t$) < 0 OR VAL(t$) > 30 THEN
    t$ = Teil$(Datum$, "Termin.Tag")
    Wiederholung% = 1
    Eingabeerfolg% = 1
END IF
LOOP UNTIL Wiederholung% = 0 OR Eingabeerfolg% = 0
IF VAL(t$) <> 0 THEN t$ = t$ + "."
COLOR 0, 7: PRINT t$;
LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
IF Eingabeerfolg% = 0 THEN GOTO DatierenSchleifeEnde
IF VAL(t$) = 0 THEN
    j$ = SPACE$(6)
    m$ = SPACE$(3)
END IF

'=====
DatumAendern:
'=====
    Termin$ = j$ + m$ + t$
    IF Antwort% MOD 2 = 1 THEN
        Aendern Daten$, Variable$ + ".Minimum", Termin$
    ELSE
        Aendern Daten$, Variable$ + ".Maximum", Termin$
    END IF
    Minimum$ = Teil$(Daten$, Variable$ + ".Minimum")
    Maximum$ = Teil$(Daten$, Variable$ + ".Maximum")
    Laenge% = SIZEOF$("Termin")
    IF Minimum$ = SPACE$(Laenge%) AND Maximum$ = SPACE$(Laenge%) THEN
        Status$ = " "
    ELSEIF HistAuswahl% <> 8 THEN
        IF Minimum$ = SPACE$(Laenge%) THEN
            Status$ = "■"
        ELSEIF Maximum$ = SPACE$(Laenge%) THEN
            Status$ = "▯"
        ELSE
            Status$ = "■"
        END IF
    END IF
    Aendern Daten$, Variable$ + ".Status", Status$
ELSEIF Antwort% = 13 THEN

    'Quellenangabe ergänzen:
    '-----'
    Laenge% = SIZEOF$("NAME")
    x% = CSRLIN
    y% = POS(0) - Laenge% - 1

```

```

Quelle$ = Teil$(Daten$, "Daten.Quelle")
Text$ = "Ergänzen Sie die 1.Quellenangabe nach Möglichkeit "
Ausgabe 2, Text$, "", 0, "M", 1, 1, 2
Text$ = "um die Quelle Ihrer Vordatierung:"
Ausgabe 2, Text$, "", 0, "M", 1, 1, 3
Wiederholung% = 1
DO
    LOCATE x%, y%
    Quelle$ = Editor$(Quelle$, Laenge%, 3, Eingabeerfolg%)
    LOOP UNTIL Quelle$ <> SPACE$(Laenge%) OR Eingabeerfolg% = 0
    IF Eingabeerfolg% = 1 THEN
        Aendern Daten$, "Daten.Quelle", Quelle$
    END IF
    LOCATE 2, 1: COLOR 7, 0: PRINT SPACE$(160)
END IF
'=====
DatierenSchleifeEnde:
'=====

'Prüfen, ob die 1.Quelle gelöscht werden muß:
'-----'
Quelledazu% = 0
FOR z% = 1 TO 6
    Variable$ = "Daten.Datum[" + LTRIM$(STR$(z%)) + "].Status"
    Status$ = Teil$(Daten$, Variable$)
    SELECT CASE Status$
    CASE "■", "■", "■"
        Quelledazu% = 1
        z% = 6
    END SELECT
NEXT z%
IF Quelledazu% = 0 AND HistAuswahl% <> 8 THEN

    'keine Vorgänger angegeben:
    '-----'
    IF Teil$(Daten$, "Daten.V[1].Direkt") = " " THEN

        'Prüfen, ob bei der Namensangabe die Zeichen " = " vorkommen:
        '-----'
        IF INSTR(Teil$(Daten$, "Daten.Name"), " = ") > 0 THEN
            Quelledazu% = 1
        END IF
    END IF
END IF
IF Quelledazu% = 0 THEN
    Aendern Daten$, "Daten.Quelle", SPACE$(SIZEOF("NAME"))
    Datumsanzeige Daten$, 13, 1, HistAuswahl%
END IF

LOOP UNTIL Antwort% = 0 OR HistAuswahl% = 8
IF HistAuswahl% <> 8 THEN
    Schreiben Eingabe%, DatenLaenge%, Auswahl&, Daten$
END IF

Datieren% = Antwort%
END FUNCTION 'Datieren% _____'

'====='
```

```

SUB Datumsanzeige (Daten$, z%, Farbe%, HistAuswahl%)
'=====
'
' Bearbeitung:
' 4. 9.2001 - 16.12.2002: Norbert Südland
' 18. 4.2007:           Norbert Südland
'-----
DIM Variable$      'AS STRING
DIM Datum$
DIM Text$
DIM Status$
DIM Zeile%
DIM Anzeigen%

IF z% < 13 THEN
    Variable$ = "Daten.Datum[ " + LTRIM$(STR$(INT((z% + 1) / 2))) + "]"
    Status$ = Teil$(Daten$, Variable$ + ".Status")
    IF HistAuswahl% = 8 OR Status$ = "■" THEN
        Anzeigen% = 1
    ELSE
        Anzeigen% = 0
        IF Status$ = "■" AND z% MOD 2 = 1 THEN
            Anzeigen% = 1
        END IF
        IF Status$ = "■" AND z% MOD 2 = 0 THEN
            Anzeigen% = 1
        END IF
    END IF
    IF Anzeigen% = 1 THEN
        SELECT CASE (z% - 1) MOD 2 + 1
        CASE 1
            Variable$ = Variable$ + ".Minimum"
        CASE 2
            Variable$ = Variable$ + ".Maximum"
        CASE ELSE
            Pause           'Programmierfehler
        END SELECT
        Datum$ = Teil$(Daten$, Variable$)
    ELSE
        Datum$ = SPACE$(SIZEOF$( "Termin" ))
    END IF
ELSE
    Variable$ = "Daten.Quelle"
    Datum$ = Teil$(Daten$, Variable$)
END IF
SELECT CASE z%
CASE 1
    Text$ = "Lebensanfang      (*) Minimum: "
CASE 2
    Text$ = "Lebensanfang      (*) Maximum: "
CASE 3
    Text$ = "1.Wirkungsanfang (A) Minimum: "
CASE 4
    Text$ = "1.Wirkungsanfang (A) Maximum: "
CASE 5
    Text$ = "2.Wirkungsanfang (W) Minimum: "
CASE 6

```

```

    Text$ = "2.Wirkungsanfang (W) Maximum: "
CASE 7
    Text$ = "2.Wirkungsende (X) Minimum: "
CASE 8
    Text$ = "2.Wirkungsende (X) Maximum: "
CASE 9
    Text$ = "1.Wirkungsende (E) Minimum: "
CASE 10
    Text$ = "1.Wirkungsende (E) Maximum: "
CASE 11
    Text$ = "Lebensende (+) Minimum: "
CASE 12
    Text$ = "Lebensende (+) Maximum: "
CASE 13
    Text$ = "Quellenangabe: "
END SELECT
Zeile% = INT((z% - 1) / 2) * 3 + 4 + (z% - 1) MOD 2 + 1
Ausgabe Farbe$, Text$, Datum$, 0, "M", 1, 1, Zeile%
END SUB 'Datumsanzeige _____'

'=====
FUNCTION Editor$ (Vorgabe$, Laenge%, Modus%, Eingabeerfolg%)
'=====
' `Modus%` = 1 bedeutet Zahleneingabe
' `Modus%` = 2 bedeutet Texteingabe
' `Modus%` = 3 bedeutet Zeicheneingabe
' `EingabeErfolg%` = 0 bedeutet Abbruch durch [ESC],
' `EingabeErfolg%` = 1 bedeutet gelungene Eingabe
'
' Bearbeitung:
' 4. 9.2001 - 27. 3.2003: Norbert Südland
'-----
DIM Antwort$ 'AS STRING
DIM a$
DIM p% 'AS INTEGER
DIM l%
DIM x%
DIM y%
DIM Start%
DIM Einfg%
DIM Ziffer& 'AS LONG

Eingabeerfolg% = 1

'Aktuelle Position:
'-----
p% = 1
IF Modus% > 1 THEN
    Vorgabe$ = LTRIM$(Vorgabe$)
    Vorgabe$ = RTRIM$(Vorgabe$)
END IF
l% = LEN(Vorgabe$)
IF l% < Laenge% THEN
    Vorgabe$ = Vorgabe$ + SPACE$(Laenge% - l%)
ELSE
    Vorgabe$ = LEFT$(Vorgabe$, Laenge%)
    l% = Laenge%

```

```
END IF
Antwort$ = Vorgabe$
COLOR 0, 7
x% = CSRLIN
y% = POS(0)
LOCATE x%, y%, 1, 12, 13
Start% = 1
Einfzg% = 1
DO
    LOCATE x%, y%
    PRINT Antwort$;
    LOCATE x%, y% + p% - 1, 1
    a$ = Taste$
    IF LEN(a$) = 1 THEN
        SELECT CASE a$
            CASE CHR$(0), CHR$(3) TO CHR$(7), CHR$(9), CHR$(11), CHR$(12)
                GOTO Schleifenende
            'unbenutzte Zeichen
            CASE CHR$(26), CHR$(28) TO CHR$(31), CHR$(255)
                GOTO Schleifenende
            'unbenutzte Zeichen
            CASE CHR$(8)
                IF p% > 1 THEN
                    p% = p% - 1
                    IF l% > 1 THEN
                        l% = l% - 1
                    END IF
                    MID$(Antwort$, p%, Laenge% - p%) = RIGHT$(Antwort$, Laenge% - p%)
                    MID$(Antwort$, Laenge%, 1) = " "
                END IF
            CASE CHR$(10), CHR$(13)
                a$ = CHR$(13)
                Eingabeerfolg% = 1
                GOTO Schleifenende
            CASE CHR$(27)
                'Maske zurück oder unbeschadeter Ausstieg
                IF Modus% = 1 AND LEN(LTRIM$(Antwort$)) = 0 AND p% = 1 THEN
                    Eingabeerfolg% = 0
                    GOTO Schleifenende
                END IF
                IF Antwort$ = Vorgabe$ AND l% = LEN(RTRIM$(Vorgabe$)) AND p% = 1 THEN
                    Eingabeerfolg% = 0
                    GOTO Schleifenende
                END IF
                Antwort$ = Vorgabe$
                l% = LEN(RTRIM$(Vorgabe$))
                p% = 1
            CASE "-", "+"
                IF Modus% = 1 THEN
                    IF Laenge% > 1 THEN
                        IF p% = 1 GOTO ZeichenSpeichern
                    ELSE
                        IF Laenge% = 1 THEN
                            IF a$ = "-" THEN
                                Antwort$ = LTRIM$(STR$((VAL(Antwort$) + 9) MOD 11))
                            ELSE
                                Antwort$ = LTRIM$(STR$((VAL(Antwort$) + 1) MOD 11))
                            END IF
                        END IF
                    END IF
                END IF
            END SELECT
    END IF
```

```

        Eingabeerfolg% = 0
    END IF
END IF
ELSE
    IF Modus% = 3 THEN
        GOTO ZeichnenSpeichern
    END IF
END IF
CASE "0" TO "9"
    GOTO ZeichnenSpeichern
CASE " " TO ")", "@ " TO "Z", "_", "a" TO "z", "Ç" TO "■"
    IF Modus% > 1 THEN
        GOTO ZeichnenSpeichern
    END IF
CASE CHR$(1) TO "■"
    IF Modus% = 3 THEN
        GOTO ZeichnenSpeichern
    END IF
END SELECT
END IF
IF LEN(a$) = 2 THEN
    'Steuertasten
    a$ = RIGHT$(a$, 1)
    SELECT CASE ASC(a$)
    CASE 59 TO 67
        '[ F1 ] bis [ F9 ]
        IF Modus% = 1 THEN
            a$ = CHR$(ASC("l") + ASC(a$) - 59)
            GOTO ZeichnenSpeichern
        END IF
    CASE 71
        'Position Home
        IF Laenge% > 1 THEN
            p% = 1
            Start% = 0
        ELSE
            IF Modus% = 1 AND Laenge% = 1 THEN
                Antwort$ = LTRIM$(STR$(1))
                Eingabeerfolg% = 0
            END IF
        END IF
    CASE 72
        'Cursor hoch
        IF Modus% = 1 AND Laenge% = 1 THEN
            Antwort$ = LTRIM$(STR$((VAL(Antwort$) + 9) MOD 10))
            Eingabeerfolg% = 0
        END IF
    CASE 75
        'Cursor links
        IF Laenge% > 1 THEN
            IF p% > 1 THEN
                p% = p% - 1
            END IF
            Start% = 0
        ELSE
            IF Modus% = 1 AND Laenge% = 1 THEN
                Antwort$ = LTRIM$(STR$((VAL(Antwort$) + 9) MOD 10))
                Eingabeerfolg% = 0
            END IF
        END IF
    END IF
END IF

```

```
        END IF
    END IF
CASE 77
    'Cursor rechts
    IF Laenge% > 1 THEN
        IF p% < Laenge% AND p% <= 1% THEN
            p% = p% + 1
        END IF
        Start% = 0
    ELSE
        IF Modus% = 1 AND Laenge% = 1 THEN
            Antwort$ = LTRIM$(STR$((VAL(Antwort$) + 1) MOD 11))
            Eingabeerfolg% = 0
        END IF
    END IF
CASE 79
    'Ende
    IF Laenge% > 1 THEN
        IF 1% < Laenge% THEN
            p% = 1% + 1
        ELSE
            p% = 1%
        END IF
        Start% = 0
    ELSE
        IF Modus% = 1 AND Laenge% = 1 THEN
            Antwort$ = LTRIM$(STR$(0))
            Eingabeerfolg% = 0
        END IF
    END IF
CASE 80
    'Cursor runter
    IF Modus% = 1 AND Laenge% = 1 THEN
        Antwort$ = LTRIM$(STR$((VAL(Antwort$) + 1) MOD 11))
        Eingabeerfolg% = 0
    END IF
CASE 82
    'Einfügen
    IF Laenge% > 1 THEN
        Start% = 0
        IF Einfg% = 1 THEN
            Einfg% = 0
            LOCATE x%, y% + p% - 1, 1, 0, 13
        ELSE
            Einfg% = 1
            LOCATE x%, y% + p% - 1, 1, 12, 13
        END IF
    END IF
CASE 83
    'Löschen
    IF Laenge% > 1 THEN
        Start% = 0
        MID$(Antwort$, p%, Laenge% - p%) = RIGHT$(Antwort$, Laenge% - p%)
        MID$(Antwort$, Laenge%, 1) = " "
        IF 1% > 1 THEN
            1% = 1% - 1
        END IF
    END IF
```

```

        END SELECT
    END IF
    GOTO Schleifenende

'=====
ZeichenSpeichern:
'=====
    IF Laenge% = 1 THEN
        Eingabeerfolg% = 1
        Antwort$ = a$
        a$ = CHR$(13)
    ELSE
        IF Start% = 1 THEN
            Start% = 0
            Antwort$ = SPACE$(Laenge%)
            l% = 0
        END IF
        IF Einfg% = 1 THEN
            IF l% <= Laenge% THEN
                IF l% < Laenge% THEN
                    l% = l% + 1
                END IF
                MID$(Antwort$, p%, l% - p% + 1) = a$ + MID$(Antwort$, p%, l% - p%)
            END IF
        ELSE
            MID$(Antwort$, p%, 1) = a$
            IF p% = l% AND l% < Laenge% THEN
                l% = l% + 1
            END IF
        END IF
        IF p% < Laenge% THEN
            p% = p% + 1
        END IF
    END IF
'=====
Schleifenende:
'=====
    LOOP UNTIL a$ = CHR$(13) OR Eingabeerfolg% = 0

    IF Eingabeerfolg% = 1 THEN
        IF Modus% THEN
            IF Modus% = 1 THEN
                Ziffer& = VAL(Antwort$)
                Antwort$ = LTRIM$(STR$(Ziffer&))
                Antwort$ = SPACE$(Laenge% - LEN(Antwort$)) + Antwort$
            END IF
            IF Modus% = 2 + 3 THEN
                Antwort$ = LTRIM$(Antwort$)
                Antwort$ = Antwort$ + SPACE$(Laenge% - LEN(Antwort$))
            END IF
        END IF
    END IF
    COLOR 7, 0
    LOCATE x%, y%, 0

    Editor$ = Antwort$
END FUNCTION 'Editor$ _____'

```

```

'=====
FUNCTION EinzelEingabe% (Daten$, z%, Farbe%, Eingabeerfolg%)
'=====
'
'  Bearbeitung:
'  5. 8.2001 - 4. 9.2001:    Norbert Südland, München
'  6. 9.2002 - 26. 9.2003:   Norbert Südland, Aalen
'  18. 4.2007 - 3. 4.2009:   Norbert Südland, Aalen
'-----
DIM K1$      'AS STRING
DIM K2$
DIM Variable$
DIM Puffer$
DIM Gleichzeitig$
DIM Quelle$
DIM a$
DIM l%      'AS INTEGER
DIM x%
DIM y%
DIM Zaehler%
DIM Zusatz%
DIM Laenge%
DIM Fertig%
DIM Bezug2Loeschen%
DIM Quelledazu%
DIM j%
DIM m%
DIM t%
DIM Start%
DIM Ende%
DIM f%
DIM GleichzeitigQuelle%

l% = LEN( "Daten.B[?]" )      'Zum Abschneiden von Variable$ benötigt.
SELECT CASE z%
CASE 1
    K1$ = CHR$(34) + "Name /Ereignis" + CHR$(34) + " fragt den Namen "
    K1$ = K1$ + "von Personen oder Ereignissen ab,"
    K2$ = "die in einem historischen Umfeld mit entsprechendem "
    K2$ = K2$ + "Umfeld datiert werden."
CASE 2, 5, 6, 9, 15, 17, 27, 29, 35
    K1$ = "* , A , W , X , E , + als Symbol für den entspr. Zeitpunkt "
    K1$ = K1$ + "bedeutet:"
    K2$ = "Lebens-, 1.Wirkungs-, 2.Wirkungs-Anfang, 2.Wirkungs- , "
    K2$ = K2$ + "1.Wirkungs-, Lebens-Ende"
CASE 3, 7
    K1$ = CHR$(34) + "Direkt davor?" + CHR$(34) + " fragt ab, wie genau die "
    K1$ = K1$ + "relative Datierung ist,"
    K2$ = "also, ob " + CHR$(34) + "Name /Ereignis" + CHR$(34) + " ohne "
    K2$ = K2$ + "Zeitlücke auf den " + CHR$(34) + "Vorgänger" + CHR$(34)
    K2$ = K2$ + " folgt oder nicht."
CASE 4, 8
    K1$ = CHR$(34) + "Vorgänger" + CHR$(34) + " fragt nach dem Vorgänger "
    K1$ = K1$ + "oder Vor-Ereignis von " + CHR$(34) + "Name" + CHR$(34) + "."
CASE 10, 22, 34
    K1$ = CHR$(34) + "Quelle" + CHR$(34) + " dient als Dokumentationshilfe, "
    K1$ = K1$ + "um die angegebene Bezugsstruktur"

```

```

K2$ = "anhand der historischen Quellen nachvollziehen zu können."
CASE 11, 18, 23, 30
  IF z% = 11 OR z% = 23 THEN
    K1$ = "Das Bezugsdatum zählt die Zeit ab einem Zeitpunkt bei "
    K1$ = K1$ + CHR$(34) + "Bezugsname" + CHR$(34) + "."
  ELSE
    K1$ = "Die Dauer bezieht sich auf den angegebenen Zeitpunkt bei "
    K1$ = K1$ + CHR$(34) + "Name" + CHR$(34) + "."
  END IF
  K2$ = "<,,> als Toleranz-Symbol bedeutet: Ordnungszahl, ohne, "
  K2$ = K2$ + "normale, beschr. Tol."
CASE 12 TO 14, 19 TO 21, 24 TO 26, 31 TO 33
  K1$ = "Berechnung in Jahren mit 12 Monaten à 30 Tage"
  K2$ = "Zulässiger Eingabebereich: 1"
  SELECT CASE z%
    CASE 12, 24
      K2$ = K2$ + ".Jahr bis 9999.Jahr - 0.Jahr bedeutet löschen."
    CASE 13, 25
      K2$ = K2$ + ".Monat bis 12.Monat - 0.Monat bedeutet löschen."
    CASE 14, 26
      K2$ = K2$ + ".Tag bis 30.Tag - 0.Tag bedeutet löschen."
    CASE 19, 31
      K2$ = K2$ + " Jahr bis 9999 Jahre - 0 Jahre bedeutet löschen."
    CASE 20, 32
      K2$ = K2$ + " Monat bis 12 Monate - 0 Monate bedeutet löschen."
    CASE 21, 33
      K2$ = K2$ + " Tag bis 30 Tage - 0 Tage bedeutet löschen."
  END SELECT
CASE 16, 28
  K1$ = CHR$(34) + "Bezugsname" + CHR$(34) + " fragt nach einer Person oder "
  K1$ = K1$ + " einem Ereigniszugehörig zu dem"
  K2$ = "Zeitpunkt, ab dem das angegebene Bezugsdatum gerechnet wird."
END SELECT
Ausgabe 2, K1$, "", 0, "M", 1, 1, 3
Ausgabe 2, K2$, "", 0, "M", 1, 1, 4
ZeilenAusgabe Daten$, z%, 2
x% = CSRLIN
y% = POS(0) - 1
Zusatz% = 0
SELECT CASE z%
CASE 1, 4, 8, 10, 16, 22, 28, 34
  Laenge% = SIZEOF%("NAME")
  y% = y% - Laenge% + 1
  SELECT CASE z%
    CASE 1
      Variable$ = "Daten.Name"
    CASE 4, 8
      Variable$ = "Daten.V[" + LTRIM$(STR$(z% / 4)) + "].Name"
    CASE 10
      Variable$ = "Daten.Quelle"
    CASE 16, 28
      Variable$ = "Daten.B[" + LTRIM$(STR$((z% - 16) / 12 + 1)) + "].Name"
    CASE 22, 34
      Variable$ = "Daten.B[" + LTRIM$(STR$((z% - 22) / 12 + 1)) + "].Quelle"
  END SELECT
  Puffer$ = Teil$(Daten$, Variable$)
DO
  LOCATE 25, 1: PRINT SPACE$(79);

```

```

LOCATE x%, y%
Puffer$ = Editor$(Puffer$, Laenge%, 3, Eingabeerfolg%)
IF Eingabeerfolg% = 1 AND z% = 1 AND Puffer$ = SPACE$(Laenge%) THEN
    IF Abfrage$("Eingabe löschen?", "N", Eingabeerfolg%) = "J" THEN
        z% = 39
    END IF
END IF
LOOP UNTIL Puffer$ <> SPACE$(Laenge%) OR z% = 39 OR Eingabeerfolg% = 0
IF Eingabeerfolg% = 0 THEN GOTO EinzelEingabeEnde
IF z% < 39 THEN Aendern Daten$, Variable$, Puffer$
CASE 2, 5, 6, 9, 15, 17, 27, 29
    SELECT CASE z%
    CASE 2, 6
        Variable$ = "Daten.V[ " + LTRIM$(STR$((z% - 2) / 4 + 1)) + "].ZP1"
    CASE 5, 9
        Variable$ = "Daten.V[ " + LTRIM$(STR$((z% - 5) / 4 + 1)) + "].ZP2"
    CASE 15, 27
        Variable$ = "Daten.B[ " + LTRIM$(STR$((z% - 15) / 12 + 1)) + "].ZP1"
    CASE 17, 29
        Variable$ = "Daten.B[ " + LTRIM$(STR$((z% - 17) / 12 + 1)) + "].ZP2"
    END SELECT
Puffer$ = FuerNutzer$(Teil$(Daten$, Variable$))
DO
    LOCATE x%, y%
    Puffer$ = Editor$(Puffer$, 1, 3, Eingabeerfolg%)
    IF Eingabeerfolg% = 0 THEN GOTO EinzelEingabeSchleifeEnde
    SELECT CASE Puffer$
    CASE "*", "A", "W", "X", "E", "+"
        Fertig% = 1
    CASE "┐"
        Puffer$ = "+"
        Fertig% = 1
    CASE "a", "w", "x", "e"
        Puffer$ = CHR$(ASC(Puffer$) - ASC("a") + ASC("A"))
        Fertig% = 1
    END SELECT
    IF Fertig% = 1 THEN Aendern Daten$, Variable$, FuerDatei$(Puffer$)
    IF Puffer$ = " " THEN
        Bezug2Loeschen% = 0
        SELECT CASE z%
        CASE 2
            Fertig% = 1
            Aendern Daten$, "Daten.V[1]", SPACE$(Laenge% = SIZEOF("Vorgänger"))
            Aendern Daten$, "Daten.V[2]", SPACE$(Laenge%)

            'Prüfen, ob Vordatierung vorhanden ist:
            '-----'
            Quelledazu% = 0
            FOR Zaehler% = 1 TO 6
                Variable$ = "Daten.Datum[ " + LTRIM$(STR$(Zaehler%)) + "].Status"
                Puffer$ = Teil$(Daten$, Variable$)
                SELECT CASE Puffer$
                CASE " ", " ", " "
                    Quelledazu% = 1
                    Zaehler% = 6
                END SELECT
            NEXT Zaehler%

```

```

'Prüfen, ob bei der Namensangabe die Zeichen " = " vorkommen:
'-----'
IF INSTR(Teil$(Daten$, "Daten.Name"), " = ") > 0 THEN
    Quelledazu% = 1
END IF

IF Quelledazu% = 1 THEN
    Zusatz% = 7
ELSE
    Aendern Daten$, "Daten.Quelle", SPACE$(Laenge%)
    Zusatz% = 8
END IF

CASE 6
    Fertig% = 1
    Aendern Daten$, "Daten.V[2]", SPACE$(SIZEOF%("Vorgänger"))
    Zusatz% = 3
CASE 17
    IF Teil$(Daten$, LEFT$(Variable$, 1%) + ".Toll") = " " THEN
        Fertig% = 1
        Aendern Daten$, "Daten.B[1]", SPACE$(SIZEOF%("Bezug"))
        Bezug2Loeschen% = 1
        Zusatz% = 17
    END IF
CASE 29
    IF Teil$(Daten$, LEFT$(Variable$, 1%) + ".Toll") = " " THEN
        Fertig% = 1
        Bezug2Loeschen% = 1
        Zusatz% = 5
    END IF
END SELECT
IF Bezug2Loeschen% = 1 THEN

    'Prüfen, ob Gleichzeitigkeiten existieren:
    '-----'
    Laenge% = SIZEOF%("Bezug")
    Gleichzeitig$ = Teil$(Daten$, "Daten.Gleichzeitig")
    IF Gleichzeitig$ = SPACE$(LEN(Gleichzeitig$)) THEN
        Aendern Daten$, "Daten.B[2]", SPACE$(Laenge%)
    ELSE
        Quelle$ = Teil$(Daten$, "Daten.B[2].Quelle")
        Aendern Daten$, "Daten.B[2]", SPACE$(Laenge%)
        Aendern Daten$, "Daten.B[2].Quelle", Quelle$
    END IF
END IF
END IF

'=====
EinzelEingabeSchleife1Ende:
'=====
    LOOP UNTIL Fertig% = 1 OR Eingabeerfolg% = 0
CASE 3, 7
    Variable$ = "Daten.V[" + LTRIM$(STR$((z% - 3) / 4 + 1)) + "].Direkt"
    Puffer$ = FuerNutzer$(Teil$(Daten$, Variable$))
DO
    LOCATE x%, y%
    Puffer$ = Editor$(Puffer$, 1, 3, Eingabeerfolg%)
    IF Eingabeerfolg% = 0 THEN GOTO EinzelEingabeSchleife2Ende
    SELECT CASE Puffer$
    CASE "j", "J", "n", "N"

```

```

        Fertig% = 1
        Aendern Daten$, Variable$, FuerDatei$(UCASE$(Puffer$))
    END SELECT
'=====
EinzelEingabeSchleife2Ende:
'=====
    LOOP UNTIL Fertig% = 1 OR Eingabeerfolg% = 0
    CASE 11, 18, 23, 30
        SELECT CASE z%
        CASE 11, 23
            Variable$ = "Daten.B[ " + LTRIM$(STR$((z% - 11) / 12 + 1)) + "].Tol1"
        CASE 18, 30
            Variable$ = "Daten.B[ " + LTRIM$(STR$((z% - 18) / 12 + 1)) + "].Tol2"
        END SELECT
        Puffer$ = Teil$(Daten$, Variable$)
    DO
        LOCATE x%, y%
        Puffer$ = Editor$(Puffer$, 1, 3, Eingabeerfolg%)
        IF Eingabeerfolg% = 0 THEN GOTO EinzelEingabeSchleife3Ende
        SELECT CASE Puffer$
        CASE "<", "=", "±", ">", " ", "+", "-", "≥", "≤", "≡"
            SELECT CASE Puffer$
            CASE "≤"
                Puffer$ = "<"
            CASE "≡"
                Puffer$ = "="
            CASE "+", "-"
                Puffer$ = "±"
            CASE "≥"
                Puffer$ = ">"
            END SELECT
        END SELECT
        Fertig% = 1
    END SELECT
    IF Fertig% = 1 THEN Aendern Daten$, Variable$, Puffer$
    IF Puffer$ = " " THEN
        IF z% = 11 OR z% = 23 THEN
            Variable$ = LEFT$(Variable$, 1%)
            Laenge% = SIZEOF$("Zeitpunkt ")
            Aendern Daten$, Variable$ + ".Datum", SPACE$(Laenge%)
            Aendern Daten$, Variable$ + ".ZP1", Puffer$
            Aendern Daten$, Variable$ + ".Name", SPACE$(SIZEOF$("NAME"))
            Zusatz% = 5
        ELSEIF z% = 18 OR z% = 30 THEN
            Variable$ = LEFT$(Variable$, 1%)
            IF Teil$(Daten$, Variable$ + ".Tol1") = " " THEN
                Aendern Daten$, Variable$ + ".ZP2", Puffer$
                Laenge% = SIZEOF$("NAME")
                Aendern Daten$, Variable$ + ".Quelle", SPACE$(Laenge%)
                IF z% = 18 THEN
                    Aendern Daten$, "Daten.B[2] ", SPACE$(SIZEOF$("Bezug"))
                    Zusatz% = 12
                END IF
                z% = z% - 1
                Zusatz% = Zusatz% + 2
            END IF
        END IF
        Aendern Daten$, Variable$ + ".Tol2", Puffer$
        Laenge% = SIZEOF$("Zeitpunkt ")
        Aendern Daten$, Variable$ + ".Dauer", SPACE$(Laenge%)
    
```

```

        Zusatz% = Zusatz% + 3
    END IF
END IF
'=====
EinzelEingabeSchleife3Ende:
'=====
    LOOP UNTIL Fertig% = 1 OR Eingabeerfolg% = 0
CASE 12 TO 14, 19 TO 21, 24 TO 26, 31 TO 33
    Variable$ = "Daten.B[" + LTRIM$(STR$(INT(z% / 23) + 1)) + "]"
    SELECT CASE z%
CASE 12, 24
        y% = y% - 3
        Variable$ = Variable$ + ".Datum"
        DO
            Puffer$ = Teil$(Daten$, Variable$ + ".Jahr")
            LOCATE x%, y%
            Puffer$ = Editor$(Puffer$, 4, 1, Eingabeerfolg%)
            IF Eingabeerfolg% = 1 THEN j% = VAL(Puffer$)
            LOOP UNTIL j% >= 0 OR Eingabeerfolg% = 0
            IF j% = 0 THEN
                Puffer$ = SPACE$(4)
                Aendern Daten$, Variable$ + ".JS", SPACE$(3)
            ELSE
                Aendern Daten$, Variable$ + ".JS", ".J "
            END IF
            Variable$ = Variable$ + ".Jahr"
CASE 13, 25
        y% = y% - 1
        Variable$ = Variable$ + ".Datum"
        DO
            Puffer$ = Teil$(Daten$, Variable$ + ".Monat")
            LOCATE x%, y%
            Puffer$ = Editor$(Puffer$, 2, 1, Eingabeerfolg%)
            IF Eingabeerfolg% = 1 THEN m% = VAL(Puffer$)
            LOOP UNTIL m% >= 0 AND m% <= 12 OR Eingabeerfolg% = 0
            IF m% = 0 THEN
                Puffer$ = SPACE$(2)
                Aendern Daten$, Variable$ + ".MS", SPACE$(3)
            ELSE
                Aendern Daten$, Variable$ + ".MS", ".M "
            END IF
            Variable$ = Variable$ + ".Monat"
CASE 14, 26
        y% = y% - 1
        Variable$ = Variable$ + ".Datum"
        DO
            Puffer$ = Teil$(Daten$, Variable$ + ".Tag")
            LOCATE x%, y%
            Puffer$ = Editor$(Puffer$, 2, 1, Eingabeerfolg%)
            IF Eingabeerfolg% = 1 THEN t% = VAL(Puffer$)
            LOOP UNTIL t% >= 0 AND t% <= 30 OR Eingabeerfolg% = 0
            IF t% = 0 THEN
                Puffer$ = SPACE$(2)
                Aendern Daten$, Variable$ + ".TS", SPACE$(2)
            ELSE
                Aendern Daten$, Variable$ + ".TS", ".T"
            END IF
            Variable$ = Variable$ + ".Tag"

```

```
CASE 19, 31
  y% = y% - 3
  Variable$ = Variable$ + ".Dauer"
  DO
    Puffer$ = Teil$(Daten$, Variable$ + ".Jahr")
    LOCATE x%, y%
    Puffer$ = Editor$(Puffer$, 4, 1, Eingabeerfolg%)
    IF Eingabeerfolg% = 1 THEN j% = VAL(Puffer$)
  LOOP UNTIL j% >= 0 OR Eingabeerfolg% = 0
  IF j% = 0 THEN
    Puffer$ = SPACE$(4)
    Aendern Daten$, Variable$ + ".JS", SPACE$(3)
  ELSE
    Aendern Daten$, Variable$ + ".JS", " J "
  END IF
  Variable$ = Variable$ + ".Jahr"
CASE 20, 32
  y% = y% - 1
  Variable$ = Variable$ + ".Dauer"
  DO
    Puffer$ = Teil$(Daten$, Variable$ + ".Monat")
    LOCATE x%, y%
    Puffer$ = Editor$(Puffer$, 2, 1, Eingabeerfolg%)
    IF Eingabeerfolg% = 1 THEN m% = VAL(Puffer$)
  LOOP UNTIL m% >= 0 AND m% <= 12 OR Eingabeerfolg% = 0
  IF m% = 0 THEN
    Puffer$ = SPACE$(2)
    Aendern Daten$, Variable$ + ".MS", SPACE$(3)
  ELSE
    Aendern Daten$, Variable$ + ".MS", " M "
  END IF
  Variable$ = Variable$ + ".Monat"
CASE 21, 33
  y% = y% - 1
  Variable$ = Variable$ + ".Dauer"
  DO
    Puffer$ = Teil$(Daten$, Variable$ + ".Tag")
    LOCATE x%, y%
    Puffer$ = Editor$(Puffer$, 2, 1, Eingabeerfolg%)
    IF Eingabeerfolg% = 1 THEN t% = VAL(Puffer$)
  LOOP UNTIL t% >= 0 AND t% <= 30 OR Eingabeerfolg% = 0
  IF t% = 0 THEN
    Puffer$ = SPACE$(2)
    Aendern Daten$, Variable$ + ".TS", SPACE$(2)
  ELSE
    Aendern Daten$, Variable$ + ".TS", " T "
  END IF
  Variable$ = Variable$ + ".Tag"
END SELECT
IF Eingabeerfolg% = 1 THEN Aendern Daten$, Variable$, Puffer$
CASE 35
  Gleichzeitig$ = FuerNutzer$(Teil$(Daten$, "Daten.Gleichzeitig"))
  y% = y% - 2
  Start% = 1
  Ende% = 3
  DO
    LOCATE x%, y%
    Laenge% = Ende% - Start% + 1
```

```

Puffer$ = MID$(Gleichzeitig$, Start%, Laenge%)
Puffer$ = Editor$(Puffer$, Laenge%, 3, Eingabeerfolg%)
IF Eingabeerfolg% = 1 THEN
  FOR f% = 1 TO Laenge%
    a$ = MID$(Puffer$, f%, 1)
    SELECT CASE a$
      CASE "*" , "A" , "W" , "X" , "E" , "+"
      CASE "+"
        MID$(Puffer$, f%, 1) = "+"
      CASE "a" , "w" , "x" , "e"
        MID$(Puffer$, f%, 1) = UCASE$(a$)
    END SELECT
  NEXT f%
  MID$(Gleichzeitig$, Start%, Laenge%) = Puffer$
  Gleichzeitig$ = GleichzeitigTest$(Gleichzeitig$)
  IF LEFT$(Gleichzeitig$, 1) = " " THEN
    Fertig% = 1
  ELSE
    IF Start% = 1 THEN
      Start% = 4: Ende% = 5: y% = y% + 6
      Aendern Daten$, "Daten.Gleichzeitig", FuerDatei$(Gleichzeitig$)
      ZeilenAusgabe Daten$, 35, 1
    ELSE
      Fertig% = 1
    END IF
  END IF
END IF
LOOP UNTIL Fertig% = 1 OR Eingabeerfolg% = 0
IF Eingabeerfolg% = 1 THEN
  Aendern Daten$, "Daten.Gleichzeitig", FuerDatei$(Gleichzeitig$)
  IF Gleichzeitig$ = SPACE$(5) THEN

    'Prüfen, ob die 3.Quelle gelöscht werden muß:
    '-----'
    IF Teil$(Daten$, "Daten.B[2].Toll1") = " " THEN
      IF Teil$(Daten$, "Daten.B[2].ZP2") = " " THEN
        Quelle$ = SPACE$(SIZEOF$("Daten.B[2].Quelle"))
        Aendern Daten$, "Daten.B[2].Quelle", Quelle$
        ZeilenAusgabe Daten$, 34, 1
      END IF
    END IF
  ELSE
    GleichzeitigQuelle% = 1
  END IF
  Zusatz% = 1
END IF
END SELECT
IF Eingabeerfolg% = 0 THEN GOTO EinzelEingabeEnde
LOCATE 3, 1
PRINT SPACE$(160);
IF z% < 39 THEN
  FOR x% = z% TO z% + Zusatz%
    ZeilenAusgabe Daten$, x%, 1
  NEXT x%
  z% = x% - 1
END IF
IF GleichzeitigQuelle% = 1 THEN z% = 38

```

```

'=====
EinzelEingabeEnde:
'=====
    EinzelEingabe% = z%
END FUNCTION 'EinzelEingabe% _____

'=====
FUNCTION Ereignis% (Symbol$)
'=====
    ,
    ' Bearbeitung:
    ' 5. 8.2001 - 4. 9.2001: Norbert Südland, München
    '-----
    DIM erg%

    SELECT CASE Symbol$
    CASE "*"
        erg% = 1
    CASE "A"
        erg% = 2
    CASE "W"
        erg% = 3
    CASE "X"
        erg% = 4
    CASE "E"
        erg% = 5
    CASE "+"
        erg% = 6
    END SELECT
    Ereignis% = erg%
END FUNCTION 'Ereignis% _____

'=====
FUNCTION FuerDatei$ (Symbol$)
'=====
    ' Wandelt 'Symbol$' in einen internationalen Kode für die Daten-Datei.
    ' Diese Funktion arbeitet auch mit alten Daten-Dateien korrekt.
    ,
    ' Bearbeitung:
    ' 20.11.2007 - 22.11.2007: Norbert Südland, D-73431 Aalen
    ,
    ' Übersetzung:
    ' 25. 3.2009: Norbert Südland, D-73431 Aalen
    '-----
    DIM Puffer$
    DIM Zaehler%

    Puffer$ = SPACE$(LEN(Symbol$))
    FOR Zaehler% = 1 TO LEN(Symbol$)
        SELECT CASE MID$(Symbol$, Zaehler%, 1)
        CASE "1", "J", "Y"
            MID$(Puffer$, Zaehler%, 1) = "1"
        CASE "0", "N"
            MID$(Puffer$, Zaehler%, 1) = "0"
        CASE "{", "*"
            MID$(Puffer$, Zaehler%, 1) = "{"
        CASE "[", "A", "B"
            MID$(Puffer$, Zaehler%, 1) = "1.Wirkungsanfang

```

```

        MID$(Puffer$, Zaehler%, 1) = "[ "
CASE "(" , "W", "α" '2.Wirkungsanfang
        MID$(Puffer$, Zaehler%, 1) = "( "
CASE ")" , "X", "Ω" '2.Wirkungsende
        MID$(Puffer$, Zaehler%, 1) = ") "
CASE "]" , "E" '1.Wirkungsende
        MID$(Puffer$, Zaehler%, 1) = "]"
CASE "}" , "+" 'Lebensende
        MID$(Puffer$, Zaehler%, 1) = "}"
CASE ELSE
        MID$(Puffer$, Zaehler%, 1) = SPACE$(1)
END SELECT
NEXT Zaehler%

FuerDatei$ = Puffer$
END FUNCTION 'FuerDatei$ _____

'=====
FUNCTION FuerNutzer$ (Symbol$)
'=====
' Wandelt die internationale 'Symbol'-Kodierung in deutschen Nutzer-Kode.
' Diese Funktion arbeitet auch mit allen alten Daten-Dateien korrekt.
'
' Bearbeitung:
' 20.11.2007 - 22.11.2007: Norbert Südland, D-73431 Aalen
'
' Übersetzung:
' 14. 2.2009 - 25. 3.2009: Norbert Südland, D-73431 Aalen
'-----
DIM Puffer$
DIM Zaehler%

Puffer$ = SPACE$(LEN(Symbol$))
FOR Zaehler% = 1 TO LEN(Symbol$)
    SELECT CASE MID$(Symbol$, Zaehler%, 1)
CASE "1", "J", "Y"
        MID$(Puffer$, Zaehler%, 1) = "J" 'Ja
CASE "0", "N"
        MID$(Puffer$, Zaehler%, 1) = "N" 'Nein
CASE "{", "*"
        MID$(Puffer$, Zaehler%, 1) = "*" 'Geburt
CASE "[", "A", "B"
        MID$(Puffer$, Zaehler%, 1) = "A" '1.Wirkungsanfang
CASE "(" , "W", "α"
        MID$(Puffer$, Zaehler%, 1) = "W" '2.Wirkungsanfang
CASE ")" , "X", "Ω"
        MID$(Puffer$, Zaehler%, 1) = "X" '2.Wirkungsende
CASE "]" , "E"
        MID$(Puffer$, Zaehler%, 1) = "E" '1.Wirkungsende
CASE "}" , "+"
        MID$(Puffer$, Zaehler%, 1) = "+" 'Lebensende
CASE ELSE
        MID$(Puffer$, Zaehler%, 1) = SPACE$(1)
    END SELECT
NEXT Zaehler%

FuerNutzer$ = Puffer$
END FUNCTION 'FuerNutzer$ _____

```

```

'=====
FUNCTION GleichzeitigTest$ (Text$)
'=====
' `Text$` muß 5 Zeichen lang sein!
'
' Bearbeitung:
' 4. 9.2001 - 24.12.2002: Norbert Südland
'-----
DIM a$(2) 'AS STRING
DIM leer$
DIM z% 'AS INTEGER
DIM y%
DIM x%(2)
DIM E%

FOR z% = 1 TO 2
  a$(z%) = SPACE$(6)
  x%(z%) = 0
  FOR y% = 1 + (z% - 1) * 3 TO 3 + (z% - 1) * 2
    E% = Ereignis$(MID$(Text$, y%, 1))
    IF E% <> 0 THEN
      IF MID$(a$(z%), E%, 1) = " " THEN
        MID$(a$(z%), E%, 1) = MID$(Text$, y%, 1)
        x%(z%) = x%(z%) + 1
      END IF
    END IF
  NEXT y%

  'Wurde nur ein Zeichen eingegeben?
  '-----
  IF x%(z%) < 2 THEN
    a$(z%) = SPACE$(6)
    x%(z%) = 0
  END IF
NEXT z%

'Reduzierung auf passende Größe:
'-----
FOR z% = 1 TO 2
  a$(z%) = RTRIM$(LTRIM$(a$(z%)))
  IF LEN(a$(z%)) > 0 THEN
    IF MID$(a$(z%), 2, 1) = " " THEN 'mind. ein Zeichen weiter rechts!
      a$(z%) = LEFT$(a$(z%), 1) + LTRIM$(RIGHT$(a$(z%), LEN(a$(z%)) - 1))
    END IF
    IF LEN(a$(z%)) > 2 THEN
      IF MID$(a$(z%), 3, 1) = " " THEN
        a$(z%) = LEFT$(a$(z%), 2) + LTRIM$(RIGHT$(a$(z%), LEN(a$(z%)) - 2))
      END IF
    END IF
  END IF
  a$(z%) = a$(z%) + SPACE$((4 - z%) - LEN(a$(z%)))
NEXT z%

'Vergleich der Kombinationen:
'-----
'Es sind 8 Koppelungen von Ereignissen prinzipiell möglich:

```

```
' *AW", "XE+", "A", "AW", "WE", "XE", "X+", "E".
```

```
'Folgende Paare der Kombination sind zulässig:
```

```
'
```

	*A	AW	WE	XE	X+	E+
*A	O	O	X	X	X	-
AW	O	O	-	-	X	-
WE	X	-	O	-	X	-
XE	X	-	-	O	O	O
X+	X	X	X	O	O	O
E+	-	-	-	O	O	O

```
'
```

```
'Die eingetragenen Symbole bedeuten:
```

```
' O Möglichkeit der Zusammenfassung zu etwas Sinnvollem
```

```
' X zulässige Kombination
```

```
' - unzulässige Kombination oder Zusammenfassung
```

```
'-----'
```

```
leer$ = SPACE$(2)
```

```
SELECT CASE a$(1)
```

```
CASE " *AW", "XE+" 'Umstellungen der Reihenfolge sind eher verwirrend!
```

```
 a$(2) = leer$
```

```
CASE " *A "
```

```
 SELECT CASE a$(2)
```

```
 CASE "AW"
```

```
 a$(1) = " *AW": a$(2) = leer$
```

```
 CASE "WE", "XE", "X+"
```

```
 CASE ELSE
```

```
 a$(2) = leer$
```

```
 END SELECT
```

```
CASE "AW "
```

```
 SELECT CASE a$(2)
```

```
 CASE " *A "
```

```
 a$(1) = " *AW": a$(2) = leer$
```

```
 CASE "X+"
```

```
 CASE ELSE
```

```
 a$(2) = leer$
```

```
 END SELECT
```

```
CASE "WE "
```

```
 SELECT CASE a$(2)
```

```
 CASE " *A "
```

```
 a$(1) = " *A ": a$(2) = "WE"
```

```
 CASE "X+"
```

```
 CASE ELSE
```

```
 a$(2) = leer$
```

```
 END SELECT
```

```
CASE "XE "
```

```
 SELECT CASE a$(2)
```

```
 CASE " *A "
```

```
 a$(1) = " *A ": a$(2) = "XE"
```

```
 CASE "X+", "E+"
```

```
 a$(1) = "XE+": a$(2) = leer$
```

```
 CASE ELSE
```

```
 a$(2) = leer$
```

```
 END SELECT
```

```
CASE "X+ "
```

```
 SELECT CASE a$(2)
```

```
 CASE " *A "
```

```
 a$(1) = " *A ": a$(2) = "X+ "
```

```

CASE "AW"
  a$(1) = "AW ": a$(2) = "X+"
CASE "WE"
  a$(1) = "WE ": a$(2) = "X+"
CASE "XE", "E+"
  a$(1) = "XE+": a$(2) = leer$
CASE ELSE
  a$(2) = leer$
END SELECT
CASE "E+ "
  SELECT CASE a$(2)
CASE "XE", "X+"
  a$(1) = "XE+": a$(2) = leer$
CASE ELSE
  a$(2) = leer$
END SELECT
CASE ELSE
  SELECT CASE a$(2)
CASE "*A", "AW", "WE", "XE", "X+", "E+"
  a$(1) = a$(2) + " ": a$(2) = leer$
CASE ELSE
  a$(2) = leer$: a$(1) = a$(2) + " "
END SELECT
END SELECT

'Ergebnis:
'-----'
GleichzeitigTest$ = a$(1) + a$(2)
END FUNCTION 'GleichzeitigTest$ _____'

'=====
FUNCTION Laden$ (Datei%, DatensatzLaenge%, Position&)
'=====
' Lädt den `Position`-ten Eintrag einer Binär-`Datei%`, deren
' `DatensatzLaenge%` konstant ist.
' Der nullte Eintrag dient der Dokumentation.
'
' Bearbeitung:
' 21. 1.2003: Norbert Südland
'-----'
DIM Daten$ 'AS STRING

Daten$ = SPACE$(DatensatzLaenge%) 'Platz schaffen!
GET #Datei%, Position& * DatensatzLaenge% + 1, Daten$

Laden$ = Daten$
END FUNCTION 'Laden$ _____'

'=====
SUB LINEINPUT (Datei%, Zeile$)
'=====
' Ersetzt den Befehl LINE INPUT, um auch unter der DOS-Box 0.73 zu laufen.
' Diese DOS-Box erzeugt mit DIR eine Ausgabe wie unter Unix und kann mit
' LINEINPUT statt mit LINE INPUT ausgewertet werden.
' Die 'Datei%' muss dazu als BINARY ACCESS READ (WRITE) geöffnet sein.
' Am Ende wird eine Zeile mit 512 Zeichen zurück gegeben, damit jede Zeile,
' die nicht mit CHR$(10) endet, noch erfasst wird. Dieser Kompromiss sorgt
' für Geschwindigkeit und reicht hier für das Auswerten von DIR aus.

```

```

'
' Bearbeitung:
' 11. 8.2009: Norbert Südland, D-73431 Aalen
' Überprüfung:
' 13. 8.2009: Norbert Südland, D-73431 Aalen
'-----'
DIM AltePosition&
DIM Laenge&
DIM NeuePosition&

Zeile$ = SPACE$(512)
AltePosition& = SEEK(Datei%)
GET #Datei%, AltePosition&, Zeile$
Laenge& = INSTR(Zeile$, CHR$(10))
IF Laenge& <> 0 THEN
    Zeile$ = LEFT$(Zeile$, Laenge& - 1)
    IF INSTR(RIGHT$(Zeile$, 1), CHR$(13)) <> 0 THEN
        Zeile$ = LEFT$(Zeile$, LEN(Zeile$) - 1)
    END IF
END IF

IF EOF(Datei%) = 0 OR Laenge& <> 0 THEN
    NeuePosition& = AltePosition& + Laenge&
    SEEK #Datei%, NeuePosition&
END IF
'PRINT LEN(Zeile$), Zeile$
'Pause
END SUB 'LINEINPUT _____'

'=====
FUNCTION Namenseintrag$ (Position&)
'=====
' Liest den `Position&`-ten Namenseintrag aus der `Eingabe%-Datei.
' Der nullte Datei-Eintrag dient der Dokumentation.
'
' Bearbeitung:
' 17.12.2002 - 21. 1.2003: Norbert Südland
'-----'
DIM Daten$ 'AS STRING

Daten$ = SPACE$(DatenLaenge%)
701 GET #Eingabe%, Position& * DatenLaenge% + 1, Daten$
Namenseintrag$ = Teil$(Daten$, "Daten.Name")
END FUNCTION 'Namenseintrag$ _____'

'=====
FUNCTION Nummer% (Text$, Start%, Ende%)
'=====
' Sucht nach Datenstruktur `Text$` und gibt Fundort zurück.
' Die Liste `GVName$()` muß sortiert sein.
'
' Bearbeitung:
' 5. 8.2001 - 4. 9.2001: Norbert Südland, München
' 18.12.2002: Norbert Südland, Aalen
'-----'
DIM Gefunden% 'AS INTEGER
DIM Anfang%

```

```

DIM Schluss%
DIM Mitte%

'Vorbereitung:
'-----'
Gefunden% = 0
Anfang% = Start%
Schluss% = Ende%

'Neuen Suchbereich ermitteln:
'-----'
WHILE Gefunden% = 0 AND Anfang% <= Schluss%
    Mitte% = INT(Anfang% / 2 + Schluss% / 2)
    SELECT CASE GVName$(Mitte%)
        CASE IS = Text$
            Gefunden% = Mitte%
        CASE IS < Text$
            Anfang% = Mitte% + 1
        CASE IS > Text$
            Schluss% = Mitte% - 1
    END SELECT
WEND

'Ergebnis:
'-----'
Nummer% = Gefunden%
END FUNCTION 'Nummer% _____'

'=====
SUB Pause
'=====
' Blendet in Zeile 25 eine Ansage ein und wartet auf einen Tastendruck.
'
' Bearbeitung:
' 18. 8.2001 Norbert Südland und Eckhard Walter, Adelshofen
'
' Überprüfung:
' 18. 8.2001 Norbert Südland und Eckhard Walter, Adelshofen
'-----'

DIM x%      'AS INTEGER
DIM y%      'AS INTEGER
DIM Antwort$ 'AS STRING

'Aktuelle Cursor-Position ermitteln:
'-----'
x% = CSRLIN
y% = POS(0)

'Zeile 25 löschen und beschriften:
'-----'
LOCATE 25, 1, 0
PRINT SPACE$(80);
Ausgabe 1, "", "Weiter mit Tastendruck", 0, "M", 1, 1, 25

'Tastendruck abwarten:
'-----'

```

```

Antwort$ = Taste$

'Zeile 25 erneut löschen:
'-----'
LOCATE 25, 1, 1
PRINT SPACE$(80);

'Cursor-Position restaurieren:
'-----'
LOCATE x%, y%
END SUB 'Pause _____'

'=====
SUB Restklasse (Vorzeichen%, Jahr%, Monat%, Tag%)
'=====
'
'  Bearbeitung:
'  4. 8.2001 - 27. 3.2003:    Norbert Südland
'-----'
DIM Tage&      'AS LONG

Tage& = Vorzeichen% * Jahr% * 360& + Monat% * 30& + Tag%
Jahr% = INT(Tage& / 360&)
Tage& = Tage& - Jahr% * 360&
Monat% = INT(Tage& / 30&)
Tag% = Tage& - Monat% * 30&
IF Jahr% < 0 THEN
    Jahr% = -Jahr%
    Vorzeichen% = -1
ELSE
    Vorzeichen% = 1
END IF
IF ABS(Jahr%) >= 10000 THEN ERROR 106
END SUB 'Restklasse _____'

'=====
SUB Schreiben (Datei%, DatensatzLaenge%, Position&, Daten$)
'=====
'  Bringt die `Daten$` auf die `DatensatzLaenge%` und schreibt sie an
'  die entsprechende `Position&` einer Binär-`Datei%` mit konstanter
'  `DatensatzLaenge%`.
'  Der nullte Datei-Eintrag dient der Dokumentation.
'
'  Bearbeitung:
'  21. 1.2003: Norbert Südland
'-----'
DIM Puffer$      'AS STRING

Puffer$ = LEFT$(Daten$, DatensatzLaenge%)
Puffer$ = SPACE$(DatensatzLaenge% - LEN(Puffer$)) + Puffer$
PUT #Datei%, Position& * DatensatzLaenge% + 1, Puffer$
END SUB 'Schreiben _____'

'=====
FUNCTION SIZEOF% (Variable$)
'=====
'  Gibt die Länge der Datenstruktur 'Variable$' zurück.

```

```

'
'  Bearbeitung:
'  4. 9.2001: Norbert Südland, München
'  6. 9.2002: Norbert Südland, Aalen
'-----'
DIM Suche$      'AS STRING
DIM Weitersuche$
DIM Laenge%     'AS INTEGER
DIM p%
DIM Ende%
DIM Wo%
DIM Ergebnis%

Laenge% = LEN(Variable$)
p% = STRLEN%(Variable$, ".")
IF p% = Laenge% THEN
    Suche$ = Variable$
ELSE
    p% = p% + 1
    p% = p% + STRLEN%(MID$(Variable$, p% + 1, Laenge% - p%), ".")
    IF p% < Laenge% THEN
        Suche$ = LEFT$(Variable$, p%)
    ELSE
        Suche$ = Variable$
    END IF
END IF
Ende% = GlobVarAnzahl%

Wo% = Nummer%(Suche$, 1, Ende%)
IF Wo% = 0 THEN
    PRINT
    PRINT "Die Datenstruktur " + CHR$(34) + Variable$; CHR$(34) + " fehlt "
801 ERROR 9
ELSE
    Ergebnis% = GVLaeenge%(Wo%)
    IF p% < Laenge% THEN
        Weitersuche$ = GVTyp$(Wo%) + MID$(Variable$, p% + 1, Laenge% - p%)
        Ergebnis% = SIZEOF%(Weitersuche$)
    END IF
END IF

SIZEOF% = Ergebnis%
END FUNCTION 'Sizeof% _____'

'=====
SUB Startbild (Antwort%, Eingabeerfolg%)
'=====
'  Gibt das Startbild aus und steuert das zugehörige Balkenmenü:
'
'  Bearbeitung:
'  4. 9.2001 - 14. 9.2001: Norbert Südland, München
'  31. 8.2002 - 4.12.2002: Norbert Südland, Aalen
'  20. 1.2009:      Norbert Südland, Aalen
'  18. 8.2012:      Norbert Südland, Aalen
'-----'
DIM Text$      'AS STRING
DIM erg$

```

```

DIM z%           'AS INTEGER
DIM Vorher%

CLS
Ausgabe 1, "H I S T O R I K", "", 0, "M", 1, 1, 1
Text$ = "Programm zur Erstellung einer Zeittafel"
Ausgabe 1, Text$, "", 0, "M", 1, 1, 2
Text$ = "Berechnung in Jahren mit 12 Monaten à 30 Tage"
Ausgabe 1, Text$, "", 0, "M", 1, 1, 3
Ausgabe 1, VersionErstellDatum$, "", 0, "M", 1, 1, 4
Text$ = "(C) 1985-" + RIGHT$(VersionErstellDatum$, 4)
Text$ = Text$ + " Norbert Südland, Eckhard Walter"
Ausgabe 1, Text$, "", 0, "M", 1, 1, 5
Text$ = "Das heutige Datum (Gregorianische Kalenderreform der christlichen"
Text$ = Text$ + " Zeitrechnung):"
Ausgabe 1, Text$, "", 0, "M", 1, 1, 7
Text$ = AktuellesDatum$
Ausgabe 1, Text$, "", 0, "M", 1, 1, 8

'Balkenmenü:
'-----'
FOR z% = 1 TO 10
    Text$ = Starttext$(z% MOD 10)
    IF (z% - Antwort%) MOD 10 = 0 THEN
        Ausgabe 1, "", Text$, 0, "M", 1, 1, 9 + z%
    ELSE
        Ausgabe 1, Text$, "", 0, "M", 1, 1, 9 + z%
    END IF
NEXT z%

'Schluss-Zeile:
'-----'
IF ENVIRON$("COMSPEC") = "Z:\COMMAND.COM" THEN
    Text$ = " [ Strg ]-[ Rollen ] "      'DOSBox 0.74
ELSE
    Text$ = " [ Strg ]-[ Pause ] "
END IF
Ausgabe 2, "Programmkode ansehen:", Text$, 0, "M", 1, 1, 25

DO
    Text$ = "Ihre Wahl: "
    Ausgabe 1, Text$, "", 0, "M", 1, 1, 21
    Eingabeerfolg% = 1
    Vorher% = Antwort%
    erg$ = Editor$(LTRIM$(STR$(Antwort%)), 1, 1, Eingabeerfolg%)
    Antwort% = VAL(erg$)
    IF Eingabeerfolg% = 0 AND Antwort% = Vorher% THEN
        IF Antwort% > 0 THEN Antwort% = (Antwort% + 1) MOD 10
        erg$ = LTRIM$(STR$(Antwort%))
    END IF
    PRINT RIGHT$(erg$, 1)
    Text$ = " " + Starttext$(Vorher%) + " "
    IF Vorher% = 0 THEN Vorher% = 10
    Ausgabe 1, Text$, "", 0, "M", 1, 1, 9 + Vorher%
    Text$ = Starttext$(Antwort%)
    IF Antwort% = 0 THEN Antwort% = 10
    Ausgabe 1, "", Text$, 0, "M", 1, 1, 9 + Antwort%

```

```

    Antwort% = Antwort% MOD 10
    LOOP UNTIL Eingabeerfolg% = 1 OR (Vorher% = 10 AND Antwort% = 0)
END SUB 'Startbild _____

'=====
FUNCTION Starttext$ (Antwort%)
'=====
'
'  Bearbeitung:
'  4. 9.2001 - 8. 9.2001: Norbert Südland, München
'-----
DIM Text$

SELECT CASE Antwort% MOD 10
CASE 1
    Text$ = "(1)...Aktuelles Datum ändern      "
CASE 2
    Text$ = "(2)...Datei anlegen /erweitern    "
CASE 3
    Text$ = "(3)...Eingabe löschen             "
CASE 4
    Text$ = "(4)...Eingabe vordatieren         "
CASE 5
    Text$ = "(5)...Übersicht und Korrektur     "
CASE 6
    Text$ = "(6)...Berechnung                  "
CASE 7
    Text$ = "(7)...Datei sortieren             "
CASE 8
    Text$ = "(8)...Ergebnis ansehen /umdatieren"
CASE 9
    Text$ = "(9)...Datei ausdrucken            "
CASE 0
    Text$ = "(0)...Programm beenden            "
END SELECT
Starttext$ = Text$
END FUNCTION 'Starttext$ _____

'=====
SUB StatusNeu (Datum$)
'=====
'
'  Bearbeitung:
'  4. 9.2001: Norbert Südland, München
'-----
DIM Status$      'AS STRING
DIM Groesse%     'AS INTEGER

SELECT CASE Teil$(Datum$, "Zeitangabe.Status")
CASE " ", " ", " " 'keine Änderung bei Vordatierung!
CASE ELSE
    Groesse% = SIZEOF("Termin")
    IF Teil$(Datum$, "Zeitangabe.Minimum") = SPACE$(Groesse%) THEN
        Status$ = "≤"
    ELSE
        Status$ = " "
    END IF
END SELECT

```

```

END IF
IF Teil$(Datum$, "Zeitangabe.Maximum") = SPACE$(Groesse%) THEN
  IF Status$ = " " THEN
    Status$ = "≥"
  ELSE
    Status$ = " "
  END IF
END IF
Aendern Datum$, "Zeitangabe.Status", Status$
END SELECT
END SUB 'StatusNeu _____'

'=====
FUNCTION STRLEN% (Text$, EndeZeichen$)
'=====
' Ermittelt die Stringlänge von `Text$` bis zum ersten Auftreten aller
' `EndeZeichen$`, so wie es in der Programmiersprache C üblich ist.
' Kommt `Ende` nicht vor, wird die Stringlänge `LEN(Text$)` zurückgegeben.
'
' Bearbeitung:
' 4. 8.2001: Norbert Südland, München
' Überprüfung:
' 4. 8.2001: Norbert Südland, München
'-----
DIM Zwischenergebnis% 'AS INTEGER

Zwischenergebnis% = INSTR(Text$, EndeZeichen$)
IF Zwischenergebnis% = 0 THEN
  Zwischenergebnis% = LEN(Text$)
ELSE
  Zwischenergebnis% = Zwischenergebnis% - 1
END IF

STRLEN% = Zwischenergebnis%
END FUNCTION 'STRLEN% _____'

'=====
FUNCTION Taste$
'=====
' Wartet auf einen Tastendruck und liefert das entsprechende ASCII-Zeichen.
'
' Bearbeitung:
' 18. 8.2001: Norbert Südland und Eckhard Walter, Adelshofen
' Überprüfung:
' 18. 8.2001: Norbert Südland und Eckhard Walter, Adelshofen
' 6. 9.2002: Norbert Südland, Aalen
'-----
DIM Antwort$ 'AS STRING

'Tastaturpuffer löschen:
'-----
WHILE INKEY$ <> ""
WEND

'Tastatur erneut abfragen, bis ein Zeichen geschrieben wurde:
'-----

```

```

Antwort$ = ""
WHILE LEN(Antwort$) = 0
    Antwort$ = INKEY$
WEND

'Ergebnis:
'-----'
Taste$ = Antwort$
END FUNCTION 'Taste$ _____'

'=====
FUNCTION Teil$ (Daten$, Variable$)
'=====
' Schneidet den über `Variable$` bezeichneten Teil von `Daten$` heraus.
'
' Bearbeitung:
' 4. 9.2001: Norbert Südland, München
' 18.12.2002: Norbert Südland, Aalen
'-----'

DIM Suche$      'AS STRING
DIM erg$
DIM Weitersuche$
DIM Laenge%     'AS INTEGER
DIM p%
DIM Ende%
DIM Wo%

erg$ = Daten$
Weitersuche$ = Variable$
DO
    Laenge% = LEN(Weitersuche$)
    p% = STRLEN$(Weitersuche$, ".")
    IF p% = Laenge% THEN
        Suche$ = Weitersuche$
    ELSE
        p% = p% + 1
        p% = p% + STRLEN$(MID$(Weitersuche$, p% + 1, Laenge% - p%), ".")
        IF p% < Laenge% THEN
            Suche$ = LEFT$(Weitersuche$, p%)
        ELSE
            Suche$ = Weitersuche$
        END IF
    END IF
    Wo% = Nummer$(Suche$, 1, GlobVarAnzahl%)
    IF Wo% = 0 THEN
        erg$ = ""
    ELSE
        erg$ = MID$(erg$, GVAnfang%(Wo%), GVLaeenge%(Wo%))
        IF p% < Laenge% THEN
            Weitersuche$ = GVTyp$(Wo%) + MID$(Weitersuche$, p% + 1, Laenge% - p%)
        END IF
    END IF
LOOP UNTIL p% = Laenge% OR Wo% = 0

Teil$ = erg$
END FUNCTION 'Teil$ _____'

```

```

'=====
SUB TerminZuZahl (Termin$, Vorzeichen%, Jahr%, Monat%, Tag%)
'=====
IF Teil$(Termin$, "Termin.VZ") = "-" THEN
    Vorzeichen% = -1
ELSE
    Vorzeichen% = 1
END IF
Jahr% = VAL(Teil$(Termin$, "Termin.Jahr"))
Monat% = VAL(Teil$(Termin$, "Termin.Monat"))
Tag% = VAL(Teil$(Termin$, "Termin.Tag"))
END SUB 'TerminZuZahl _____

'=====
FUNCTION TextAuffuellen$ (Text$, WunschLaenge%, Zeichen$)
'=====
' Füllt den `Text$` bis zum Erreichen der `WunschLaenge%` mit `Zeichen$`
' auf, wobei eine positive `WunschLaenge%` ein linksbündiges und eine
' negative `WunschLaenge%` ein rechtsbündiges Ergebnis erzeugt.
'
' Wird als `Zeichen$` ein String der Länge Null übergeben, so bestimmt
' die Länge vom `Text$` die maximale `WunschLaenge%`.
'
' Bislang sind Probleme mit dem Speicher-Überlauf bei extrem langen
' Zeichenketten nicht abgefangen, da sie von der Konfiguration des
' aufrufenden BASIC-Interpreters abhängen!
'
' Bearbeitung:
' 18. 8.2001 Norbert Südland, Adelshofen
'
' Überprüfung:
' 18. 8.2001 Norbert Südland, Adelshofen
'-----
DIM TextKopie$      'AS STRING
DIM Ergebnis$       'AS STRING

'Vorbereitung:
'-----
Ergebnis$ = Text$

'Auffüllen des Textes:
'-----
IF WunschLaenge% < 0 THEN

    'Rechtsbündig:
    '-----
    IF LEN(Zeichen$) > 0 THEN
        WHILE LEN(Ergebnis$) < -WunschLaenge%
            Ergebnis$ = Zeichen$ + Ergebnis$
        WEND
    END IF
    Ergebnis$ = RIGHT$(Ergebnis$, -WunschLaenge%)
ELSE

    'Linksbündig:
    '-----
    IF LEN(Zeichen$) > 0 THEN

```

```

        WHILE LEN(Ergebnis$) < WunschLaenge%
            Ergebnis$ = Ergebnis$ + Zeichen$
        WEND
    END IF
    Ergebnis$ = LEFT$(Ergebnis$, WunschLaenge%)
END IF

'Ergebnis:
'-----'
TextAuffuellen$ = Ergebnis$
END FUNCTION 'TextAuffuellen$ _____'

'=====
FUNCTION Ueberblick& (HistAuswahl%, Vorgabe&, Eingabeerfolg%)
'=====
' Gibt einen `Überblick` über die Einträge einer Datei.
'
' Bearbeitung:
' 5. 8.2001 - 10. 2.2003:    Norbert Südland
' Überprüfung:
' 17.12.2002:                Norbert Südland, Aalen
'-----'
DIM Text1$           'AS STRING
DIM Text2$
DIM Name$
DIM a$
DIM z%               'AS INTEGER
DIM Antwort&         'AS LONG INTEGER
DIM Weiter&
DIM Start&
DIM Ende&
DIM Aktuell&

'Leere Datei abfangen:
'-----'
IF GEL& <= 0 THEN
    Antwort& = 0
    Eingabeerfolg% = 0
    GOTO UeberblickEnde
ELSE
    Aktuell& = FNMIN(FNMAX(1&, Vorgabe&), GEL&)

    'Hinweis: `CLNG( )` arbeitet bei QuickBasic 4.5 unzuverlässig!
    '-----'
    Start& = Aktuell& - ((Aktuell& - 1) MOD 57)
    Ende& = FNMIN(GEL&, Aktuell& - ((Aktuell& - 1) MOD 57) + 59&)
    IF GEL& > 60 AND GEL& - Start& < 59 THEN
        Start& = GEL& - 59
        Ende& = GEL&
    END IF
END IF
END IF

DO
    'Menü-Beschriftung:
    '-----'
    CLS
    SELECT CASE HistAuswahl%

```

```

CASE 3
    Text1$ = "(3)...Eingabe löschen"
    Text2$ = "Bitte einen Namen auswählen:"
CASE 4
    Text1$ = "(4)...Eingabe vordatieren"
    Text2$ = "Bitte einen Namen auswählen:"
CASE 5
    Text1$ = "(5)...Übersicht und Korrektur"
    Text2$ = "Bitte einen Namen auswählen:"
CASE 6
    Text1$ = "(7)...Eingabe sortieren"
    Text2$ = "Bitte einen Namen auswählen:"
CASE 7
    Text1$ = "(7)...Eingabe sortieren"
    Text2$ = "Bitte den neuen Ort auswählen:"
CASE 8
    Text1$ = "(8)...Datei umdatieren"
    Text2$ = "Bitte ein Datum zum expliziten Umdatieren aussuchen:"
END SELECT
Ausgabe 1, "", Text1$, 0, "M", 1, 1, 1
Ausgabe 2, Text2$, "", 0, "M", 1, 1, 2
Text1$ = "Zum Verlassen des Menüs [ ESC ] drücken."
Ausgabe 2, Text1$, "", 0, "M", 1, 1, 25

'Darstellungsfenster:
'-----'
FOR z% = 0 TO Ende& - Start&
    Name$ = Namenseintrag$(Start& + z%)
    Ausgabe 1, Name$, "", 0, "M", (z% MOD 3) + 1, 3, INT(z% / 3) + 4
NEXT z%

'Tastatur-Eingabe auswerten:
'-----'
Antwort& = 0
DO
    Name$ = " " + Namenseintrag$(Aktuell&) + " "
    z% = INT(Aktuell& - Start&)
    Ausgabe 2, "", Name$, 0, "M", z% MOD 3 + 1, 3, INT(z% / 3) + 4
    a$ = Taste$
    Weiter& = Aktuell&
    SELECT CASE ASC(a$)
    CASE 10, 13
        a$ = CHR$(13)
        Antwort& = Aktuell&
        Eingabeerfolg% = 1
    CASE 27
        a$ = CHR$(13)
        Antwort& = Aktuell&
        Eingabeerfolg% = 0
    END SELECT
    IF LEN(a$) = 2 THEN
        a$ = RIGHT$(a$, 1)
        SELECT CASE ASC(a$)
        CASE 71 'Home
            Weiter& = Start&
        CASE 72 'Cursor hoch
            IF Aktuell& - 3 >= Start& THEN
                Weiter& = Aktuell& - 3

```

```
ELSEIF GEL& > 60 THEN
  a$ = CHR$(13)
  Weiter& = FNMAX(1&, Aktuell& - 3)
  Start& = Start& - 3
  Ende& = Ende& - 3
  IF Start& < 1 THEN
    Weiter& = Aktuell&
    Start& = 1
    Ende& = 60
  END IF
END IF
CASE 73 'Bild hoch
  IF Start& = 1 THEN
    Weiter& = 1
  ELSEIF GEL& > 60 THEN
    a$ = CHR$(13)
    Weiter& = FNMAX(1&, Aktuell& - 57)
    Start& = Start& - 57
    Ende& = Ende& - 57
    IF Start& < 1 THEN
      Start& = 1
      Ende& = 60
    END IF
  END IF
END IF
CASE 75 'Cursor links
  IF Aktuell& > Start& THEN
    Weiter& = Aktuell& - 1
  ELSEIF GEL& > 60 THEN
    a$ = CHR$(13)
    Weiter& = FNMAX(1&, Aktuell& - 1)
    Start& = Start& - 3
    Ende& = Ende& - 3
    IF Start& < 1 THEN
      Start& = 1
      Ende& = 60
    END IF
  END IF
END IF
CASE 77 'Cursor rechts
  IF Aktuell& < Ende& THEN
    Weiter& = Aktuell& + 1
  ELSEIF GEL& > 60 THEN
    a$ = CHR$(13)
    Weiter& = FNMIN(Aktuell& + 1, GEL&)
    Start& = Start& + 3
    Ende& = Ende& + 3
    IF Ende& > GEL& THEN
      Start& = GEL& - 59
      Ende& = GEL&
    END IF
  END IF
END IF
CASE 79
  Weiter& = Ende&
CASE 80 'Cursor runter
  IF Aktuell& + 3 <= Ende& THEN
    Weiter& = Aktuell& + 3
  ELSEIF GEL& > 60 THEN
    a$ = CHR$(13)
    Weiter& = FNMIN(Aktuell& + 3, GEL&)
```

```

        Start& = Start& + 3
        Ende& = Ende& + 3
        IF Ende& > GEL& THEN
            Weiter& = Aktuell&
            Start& = GEL& - 59
            Ende& = GEL&
        END IF
    END IF
CASE 81 'Bild runter
    IF Ende& = GEL& THEN
        Weiter& = Ende&
    ELSEIF GEL& > 60 THEN
        a$ = CHR$(13)
        Weiter& = FNMIN(Aktuell& + 57, GEL&)
        Start& = Start& + 57
        Ende& = Ende& + 57
        IF Ende& > GEL& THEN
            Start& = GEL& - 59
            Ende& = GEL&
        END IF
    END IF
CASE 117, 118 '[ Strg - Ende ], [ Strg - Bild runter ]
    IF Ende& = GEL& THEN
        Weiter& = GEL&
    ELSEIF GEL& > 60 THEN
        a$ = CHR$(13)
        Weiter& = GEL&
        Start& = GEL& - 59
        Ende& = GEL&
    END IF
CASE 119, 132 '[ Strg - Pos1 ], [ Strg - Bild hoch ]
    IF Start& = 1 THEN
        Weiter& = 1
    ELSEIF GEL& > 60 THEN
        a$ = CHR$(13)
        Weiter& = 1
        Start& = 1
        Ende& = 60&
    END IF
END SELECT
Ausgabe 1, Name$, "", 0, "M", z% MOD 3 + 1, 3, INT(z% / 3) + 4
Name$ = Namenseintrag$(Weiter&)
Aktuell& = Weiter&
END IF
LOOP UNTIL a$ = CHR$(13)
LOOP UNTIL Antwort& > 0

'=====
UeberblickEnde:
'=====
    Ueberblick& = Antwort&
END FUNCTION 'Ueberblick& _____

'=====
SUB Umkehren (v1%, j1%, m1%, t1%, v2%, j2%, m2%, t2%, Richtung$)
'=====
' Wenn 'Richtung$' = "-" , wird die Zeitangabe 'Datum$' gespiegelt.
' 'Richtung' = "+" hat keine Wirkung.

```

```

'
' Bearbeitung:
' 4. 9.2001: Norbert Südland, München
'-----'
DIM z%      'AS INTEGER
DIM v%
DIM j%
DIM m%
DIM t%

IF Richtung$ = "-" THEN
  FOR z% = 1 TO 2
    IF z% = 1 THEN
      v% = v1%: j% = j1%: m% = m1%: t% = t1%
    ELSE
      v% = v2%: j% = j2%: m% = m2%: t% = t2%
    END IF
    v% = -v%
    m% = -m%
    t% = -t%
    Restklasse v%, j%, m%, t%
    IF z% = 1 THEN
      v1% = v%: j1% = j%: m1% = m%: t1% = t%
    ELSE
      v2% = v%: j2% = j%: m2% = m%: t2% = t%
    END IF
  NEXT z%
END IF
END SUB 'Umkehren _____'

'=====
FUNCTION Wochentag% (jh%, j%, m%, t%)
'=====
' Berechnet den Wochentag zu einem angegebenen Datum.
'
' Wertebereich für Jahrhundert (jh%): -10 bis 100 ohne 0
' Wertebereich für Jahreszahl (j%) : 1 bis 100
' Wertebereich für Monatszahl (m%) : 1 bis 12
' Wertebereich für Tageszahl (t%) : 1 bis 31
' Diskutabler Wertebereich fürs Datum: 1.1.-999 bis 31.12.9999
'
' Die übergebenen Zahlen werden zu einem sinnvollen Datum gemacht.
'-----'
' Die (willkürlich wählbare) Referenz wird auf Freitag, den 14. 5.1948,
' als dem Tag der Wiedergründung Israels festgesetzt.
'
' Die Gregorianische Kalenderreform mit Zeitsprung von Donnerstag, den
' 4.10.1582, auf Freitag, den 15.10.1582, wird berücksichtigt.
' ([Zem1987], Einleitung und Abschnitt 2.65, Seiten 11 und 28-29)
'
' Die Julianische Kalenderreform von 45 v.Chr. durch Caius Julius Caesar
' ([Zem1987], Abschnitt 2.64, Seite 27) wird berücksichtigt.
'
' Vor der Julianischen Kalenderreform wird das ägyptische Sonnenjahr mit
' genau 365 Tagen verwendet, das in 1460 Jahren bereits 1461 Jahre zählt.
'
' Der Vergleich mit historischen Datierungen weicht mitunter aufgrund

```

```

' dieser Besonderheiten ab und bedarf der Umrechnung.
'
' Zu Vergleichszwecken seien einige unumstrittene Datierungen angegeben:
'   Freitag, den 14. 5.1948: Wiedergründung Israels
'   Samstag, den 31.10.1517: Thesenanschlag Luthers
'   Sonntag, den  9. 4.  30: Auferstehung Jesu Christi [Rusk1974]
'
' [Rusk1974]
' Rusk R. in "Christianity Today", Washington D.C., Vol XIII, Nr. 13,
' (29. 3.1974), pp 4 (720) - 6 (722);
' Ohne Kommentar aus dem Englischen übersetzt von Erhard Kietz, Phys. i.R.
'
' [Zem1987]
' Zemanek H.: "Kalender und Chronologie", R. Oldenbourg Verlag München,
' Wien, 4. verbesserte Auflage, (1987)
'-----'
'
' Bearbeitung:
'  5. 8.2001 -  4. 9.2001:   Norbert Südland, München
'  1. 9.2002 -  5. 9.2002:   Norbert Südland, Aalen
' Überprüfung:
'  1. 9.2002 -  5. 9.2002:   Norbert Südland, Aalen
'-----'
DIM Jahrhundert%      'AS INTEGER
DIM Jahr%             'AS INTEGER
DIM Monat%           'AS INTEGER
DIM Tag%             'AS INTEGER
DIM Schalt%          'AS INTEGER
DIM Bereich%         'AS INTEGER
DIM Rest&            'AS LONG
DIM WoTag&           'AS LONG

'Vorbereitung:
'-----'
Jahrhundert% = jh%
Jahr% = j% + INT((m% - 1) / 12)      'nur Jahre mit 12 Monaten
Monat% = ((m% - 1) MOD 12) + 1      'nur 12 Monate möglich
WHILE Monat% < 1
    Monat% = Monat% + 12            'Monat muß im richtigen Bereich sein!
WEND
Tag% = t%

'Das 0.Jahrhundert und das 0.Jahr existiert nicht:
'-----'
IF Jahr% < 0 THEN Jahr% = Jahr% + 1  'Schaltjahr durch 4 glatt teilbar!
SELECT CASE Jahrhundert%
CASE IS > 0
    Jahrhundert% = Jahrhundert% - 1
    Jahr% = 100 * Jahrhundert% + Jahr%
CASE IS < 0
    Jahr% = 100 * Jahrhundert% + Jahr%
END SELECT

'Tage ab 14. 5.1948 orientieren:
'-----'
WoTag& = 6                        'Freitag, der 14. 5.1948
WoTag& = WoTag& + Tag% - 14

```

```

SELECT CASE Monat%
CASE 1
    WoTag& = WoTag& - 120
CASE 2
    WoTag& = WoTag& - 89
CASE 3
    WoTag& = WoTag& - 61
CASE 4
    WoTag& = WoTag& - 30
CASE 6
    WoTag& = WoTag& + 31
CASE 7
    WoTag& = WoTag& + 61
CASE 8
    WoTag& = WoTag& + 92
CASE 9
    WoTag& = WoTag& + 123
CASE 10
    WoTag& = WoTag& + 153
CASE 11
    WoTag& = WoTag& + 184
CASE 12
    WoTag& = WoTag& + 214
END SELECT

```

'Rechnung nach ägyptischem Sonnenjahr:

'-----'

WoTag& = WoTag& + (Jahr% - 1948) * 365&

'Julianische Schalttage ab 45 v.Chr. (Das 0.Jahr gibt es nicht):

'-----'

WoTag& = WoTag& + INT((FNMAX(-45, Jahr%) - 1948) / 4)

'Gregorianischer Ausgleich zum Jahr 1948 ab dem 15.10.1582:

'-----'

WoTag& = WoTag& - INT(FNMAX(Jahr%, 1582) / 100 - 19) 'INT(1948 / 100) = 19

WoTag& = WoTag& + INT(FNMAX(Jahr%, 1582) / 400 - 4) 'INT(1948 / 400) = 4

IF Jahr% < 1582 OR (Jahr% = 1582 AND Monat% < 10) THEN

 WoTag& = WoTag& + 10 'historische Fehltage überbrücken

END IF

IF Jahr% = 1582 AND Monat% = 10 AND Tag% < 5 THEN

 WoTag& = WoTag& + 10 'historische Fehltage überbrücken

END IF

'Im Schaltjahr vor dem Schalttag:

'-----'

IF Jahr% > -45 AND Jahr% MOD 4 = 0 AND Monat% < 3 THEN

 WoTag& = WoTag& - 1 'Schalttag findet statt

'Gregorianische Kalenderreform:

'-----'

IF Jahr% > 1582 AND Jahr% MOD 100 = 0 AND Jahr% MOD 400 <> 0 THEN

 WoTag& = WoTag& + 1 'kein Schalttag bei norm. Jh-Wende

END IF

END IF

'Datumszahlen nach Ergebnis ausrichten:

'-----'

```
SELECT CASE WoTag&
CASE IS <= -727635
```

```
'Ägyptischer Sonnenkalender bis 28.2.45 v.Chr.:
'-----'
' Ein normales Sonnenjahr dauert 365 Tage.
' Die Differenz 1.1.1948 bis Freitag, den 14. 5.1948, ergibt 128 Tage.
' Bis 1948 läßt die Gregorianische Reform 13 der 498 Schalttage aus.
' Die Wochentagszahl ist negativ, wodurch Auslassung Addition ist.
'-----'
Jahr% = INT((WoTag& + 498 - 13 + 31 + 28 + 31 + 30 + 14 - 6) / 365)
Rest& = (WoTag& + 498 - 13 + 128) - Jahr% * 365&
Jahr% = Jahr% + 1948
Schalt% = 0
```

```
CASE -727634 TO -133520
```

```
'Julianischer Kalender vom 29.2.45 v.Chr. bis 4.10.1582 n.Chr.:
'-----'
' Eine 4-Jahres-Periode dauert 1461 Tage.
' Ein normales Sonnenjahr dauert 365 Tage.
' Der letzte Block einer Periode kann eine Überlänge besitzen.
' Die Differenz 1.1.1945 bis Freitag, den 14. 5.1948, ergibt 1223 Tage.
' Dieser Übergang zur Gregorianischen Reform läßt 13 Tage aus.
'-----'
```

```
'Jahres-Quartett zuordnen:
'-----'
Bereich% = FNMIN(INT((WoTag& + 1223 - 13) / 1461), 3)
Rest& = (WoTag& + 1223 - 13) - Bereich% * 1461&
Jahr% = 1945 + 4 * Bereich%
Schalt% = 0
```

```
'Das letzte Jahr im Jahres-Quartett ist eventuell länger:
'-----'
Bereich% = FNMIN(INT(Rest& / 365), 3)
Rest& = Rest& - Bereich% * 365&
Jahr% = Jahr% + Bereich%
IF Bereich% = 3 THEN
    Schalt% = Schalt% + 1          'Schaltjahr markieren
END IF
```

```
CASE IS >= -133519
```

```
'Gregorianische Kalenderreform ab 15.10.1582 n.Chr.:
'-----'
' Eine 400-Jahres-Periode dauert 146097 Tage ([Zem1987], Seite 29).
' Eine 100-Jahres-Periode dauert 36524 Tage.
' Eine 4-Jahres-Periode dauert 1461 Tage.
' Ein normales Sonnenjahr dauert 365 Tage.
' Jeweils der letztes Block einer Periode kann eine Überlänge besitzen.
' Die Differenz 1.1.1601 bis Freitag, den 14. 5.1948, ergibt 126866 Tage.
'-----'
```

```
'Jahrhundert-Quartetts zuordnen:
'-----'
Bereich% = INT((WoTag& + 126866) / 146097)
Rest& = (WoTag& + 126866) - Bereich% * 146097
```

```
Jahr% = 1601 + 400 * Bereich%
Schalt% = 0

'Das letzte Jahrhundert im Quartett ist länger:
'-----'
Bereich% = FNMIN(INT(Rest& / 36524), 3)
Rest& = Rest& - Bereich% * 36524
Jahr% = Jahr% + 100 * Bereich%
IF Bereich% = 3 THEN
    Schalt% = 1                                'Schalt-Jahrhundert markieren
END IF

'Jedes Jahrhundert besitzt 25 Jahres-Quartetts:
'-----'
Bereich% = FNMIN(INT((Rest&) / 1461), 24)
Rest& = Rest& - Bereich% * 1461&
Jahr% = Jahr% + 4 * Bereich%
IF Bereich% = 24 AND Rest& >= 3 * 365 THEN
    Schalt% = Schalt% - 1                    'Normal kein Schaltjahr am Jh.-Ende
ELSE
    Schalt% = 0
END IF

'Das letzte Jahr im Jahres-Quartett ist eventuell länger:
'-----'
Bereich% = FNMIN(INT(Rest& / 365), 3)
Rest& = Rest& - Bereich% * 365&
Jahr% = Jahr% + Bereich%
IF Bereich% = 3 THEN
    Schalt% = Schalt% + 1                    'Schaltjahr markieren
END IF

END SELECT

'Datumszahlen zuordnen:
'-----'
IF Jahr% < 1 THEN Jahr% = Jahr% - 1
SELECT CASE Rest&
CASE 0 TO 30
    Monat% = 1
    Tag% = Rest& + 1
CASE 31 TO 58 + Schalt%
    Monat% = 2
    Tag% = Rest& - 30
CASE 59 + Schalt% TO 89 + Schalt%
    Monat% = 3
    Tag% = Rest& - 58 - Schalt%
CASE 90 + Schalt% TO 119 + Schalt%
    Monat% = 4
    Tag% = Rest& - 89 - Schalt%
CASE 120 + Schalt% TO 150 + Schalt%
    Monat% = 5
    Tag% = Rest& - 119 - Schalt%
CASE 151 + Schalt% TO 180 + Schalt%
    Monat% = 6
    Tag% = Rest& - 150 - Schalt%
CASE 181 + Schalt% TO 211 + Schalt%
    Monat% = 7
```

```

    Tag% = Rest& - 180 - Schalt%
CASE 212 + Schalt% TO 242 + Schalt%
    Monat% = 8
    Tag% = Rest& - 211 - Schalt%
CASE 243 + Schalt% TO 272 + Schalt%
    Monat% = 9
    Tag% = Rest& - 242 - Schalt%
CASE 273 + Schalt% TO 303 + Schalt%
    Monat% = 10
    Tag% = Rest& - 272 - Schalt%
CASE 304 + Schalt% TO 333 + Schalt%
    Monat% = 11
    Tag% = Rest& - 303 - Schalt%
CASE 334 + Schalt% TO 364 + Schalt%
    Monat% = 12
    Tag% = Rest& - 333 - Schalt%
CASE ELSE
    Pause                'Programmierfehler in der Wochentagsberechnung!
    PRINT Rest&
    SYSTEM
END SELECT

jh% = Jahrhundert%
IF jh% >= 0 THEN jh% = jh% + 1      'Jahrhundert-Angabe ist übergeordnet.
j% = Jahr%                        'Jahresangabe wird voll übernommen.
m% = Monat%
t% = Tag%

'Konkrete Nummer des Wochentags:
'-----'
WoTag& = WoTag& MOD 7&
WHILE WoTag& < 1
    WoTag& = WoTag& + 7&
WEND

Wochentag% = WoTag&
END FUNCTION 'Wochentag% _____'

'=====
FUNCTION ZahlZuTermin$ (Vorzeichen%, Jahr%, Monat%, Tag%)
'=====
'
'  Bearbeitung:
'    4. 9.2001: Norbert Südland, München
'-----'
DIM Termin$      'AS STRING

Termin$ = SPACE$(SIZEOF%("Termin"))
Jahr% = Vorzeichen% * Jahr%
IF Jahr% < 0 THEN
    Vorzeichen% = -1
    Jahr% = -Jahr%
ELSE
    Vorzeichen% = 1
END IF
IF Vorzeichen% = 1 THEN
    Aendern Termin$, "Termin.VZ ", " "

```

```

ELSE
    Andern Termin$, "Termin.VZ ", "- "
END IF
IF Jahr% <> 0 THEN
    Andern Termin$, "Termin.Jahr ", LTRIM$(STR$(Jahr%))
    Andern Termin$, "Termin.JS ", "."
END IF
IF Monat% <> 0 THEN
    Andern Termin$, "Termin.Monat ", LTRIM$(STR$(Monat%))
    Andern Termin$, "Termin.MS ", "."
END IF
IF Tag% <> 0 THEN
    Andern Termin$, "Termin.Tag ", LTRIM$(STR$(Tag%))
    Andern Termin$, "Termin.TS ", "."
END IF
ZahlZuTermin$ = Termin$
END FUNCTION 'ZahlZuTermin$ _____ '

'=====
SUB ZeilenAusgabe (Daten$, z%, Farbe%)
'=====
'
'  Bearbeitung:
'  4. 9.2001: Norbert Südland, München
'-----

DIM Frage$      'AS STRING
DIM Variable$
DIM Puffer$
DIM Offset%     'AS INTEGER
DIM Bereich%
DIM Bereiche%
DIM y%

Offset% = 0
SELECT CASE z%
CASE 1
    Frage$ = "Name /Ereignis: "
    Variable$ = "Daten.Name"
    Bereich% = 1
    Bereiche% = 1
    y% = 5
CASE 2, 6
    Bereich% = (z% - 2) / 4 + 1
    Bereiche% = 2
    Frage$ = "Zeitpunkt beim Vorg.: "
    Variable$ = "Daten.V[" + LTRIM$(STR$(Bereich%)) + "].ZP1"
    y% = 6
CASE 3, 7
    Bereich% = (z% - 3) / 4 + 1
    Bereiche% = 2
    Frage$ = " Direkt davor? [J/N] "
    Variable$ = "Daten.V[" + LTRIM$(STR$(Bereich%)) + "].Direkt "
    y% = 7
CASE 4, 8
    Bereich% = (z% - 4) / 4 + 1
    Bereiche% = 2
    Frage$ = LTRIM$(STR$(Bereich%)) + ".Vorgänger: "

```

```
Variable$ = "Daten.V[ " + LTRIM$(STR$(Bereich%)) + "].Name "
y% = 8
CASE 5, 9
  Bereich% = (z% - 5) / 4 + 1
  Bereiche% = 2
  Frage$ = " Zeitpunkt bei Name: "
  Variable$ = "Daten.V[ " + LTRIM$(STR$(Bereich%)) + "].ZP2 "
  y% = 9
CASE 10
  Bereich% = 1
  Bereiche% = 1
  Frage$ = "1.Quelle: "
  Variable$ = "Daten.Quelle "
  y% = 10
CASE 11, 23
  Bereich% = (z% - 11) / 12 + 1
  Bereiche% = 2
  Offset% = -1
  Frage$ = "Toleranz " + LTRIM$(STR$(Bereich%)) + ".Bezugsdatum: "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Toll "
  y% = 11
CASE 12, 24
  Bereich% = (z% - 12) / 12 + 1
  Bereiche% = 2
  Frage$ = " Welches Jahr?      "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Datum.Jahr "
  y% = 12
CASE 13, 25
  Bereich% = (z% - 13) / 12 + 1
  Bereiche% = 2
  Frage$ = " Welcher Monat?      "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Datum.Monat "
  y% = 13
CASE 14, 26
  Bereich% = (z% - 14) / 12 + 1
  Bereiche% = 2
  Frage$ = " Welcher Tag?        "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Datum.Tag "
  y% = 14
CASE 15, 27
  Bereich% = (z% - 15) / 12 + 1
  Bereiche% = 2
  Offset% = -2
  Frage$ = "Zeitpunkt bei Bezugsname: "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].ZP1 "
  y% = 15
CASE 16, 28
  Bereich% = (z% - 16) / 12 + 1
  Bereiche% = 2
  Frage$ = LTRIM$(STR$(Bereich%)) + ".Bezugsname: "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Name "
  y% = 16
CASE 17, 29
  Bereich% = (z% - 17) / 12 + 1
  Bereiche% = 2
  Frage$ = " Zeitpunkt bei Name: "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].ZP2 "
  y% = 17
```

```

CASE 18, 30
  Bereich% = (z% - 18) / 12 + 1
  Bereiche% = 2
  Frage$ = " Toleranz " + LTRIM$(STR$(Bereich%)) + ".Dauer:  "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Tol2 "
  y% = 18
CASE 19, 31
  Bereich% = (z% - 19) / 12 + 1
  Bereiche% = 2
  Frage$ = " Wieviele Jahre?  "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Dauer.Jahr "
  y% = 19
CASE 20, 32
  Bereich% = (z% - 20) / 12 + 1
  Bereiche% = 2
  Frage$ = " Wieviele Monate?  "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Dauer.Monat "
  y% = 20
CASE 21, 33
  Bereich% = (z% - 21) / 12 + 1
  Bereiche% = 2
  Frage$ = " Wieviele Tage?      "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Dauer.Tag "
  y% = 21
CASE 22, 34
  Bereich% = (z% - 22) / 12 + 1
  Bereiche% = 2
  Frage$ = "      " + LTRIM$(STR$(Bereich% + 1)) + ".Quelle:  "
  Variable$ = "Daten.B[ " + LTRIM$(STR$(Bereich%)) + "].Quelle "
  y% = 22
CASE 35, 36
  Bereich% = 1
  Bereiche% = 1
  IF z% = 35 THEN
    Frage$ = "Gleichzeitige Ereignisse:  "
    Offset% = -3
  ELSE
    Frage$ = ""
    Offset% = 15
  END IF
  Variable$ = "Daten.Gleichzeitig"
  y% = 23
END SELECT
Puffer$ = Teil$(Daten$, Variable$)
SELECT CASE z%
CASE 2, 3, 5, 6, 7, 9, 15, 17, 27, 29
  Puffer$ = FuerNutzer$(Puffer$)
CASE 35
  Puffer$ = FuerNutzer$(LEFT$(Puffer$, 3))
CASE 36
  Puffer$ = FuerNutzer$(RIGHT$(Puffer$, 2))
END SELECT
Ausgabe Farbe%, Frage$, Puffer$, Offset%, "M", Bereich%, Bereiche%, y%

END SUB 'Zeilenausgabe _____'

'=====
SUB ZeitSchieben (v1%, j1%, m1%, t1%, Status$, v2%, j2%, m2%, t2%, Richtung$)

```

```
'=====
'
'  Bearbeitung:
'  4. 9.2001: Norbert Südland, München
'-----
DIM Start%      'AS INTEGER
DIM Ende%
DIM z%
DIM v%
DIM j%
DIM m%
DIM t%

SELECT CASE Status$
CASE " "
    Start% = 1
    Ende% = 2
CASE "≥"
    Start% = 1
    Ende% = 1
CASE "≤"
    Start% = 2
    Ende% = 2
END SELECT
FOR z% = Start% TO Ende%
    IF z% = 1 THEN
        v% = v1%: j% = j1%: m% = m1%: t% = t1%
    ELSE
        v% = v2%: j% = j2%: m% = m2%: t% = t2%
    END IF
    IF Richtung$ = "+" THEN
        Restklasse v%, j%, m%, t%
        IF v% = 1 THEN
            j% = j% + 1
        END IF
        m% = m% + 1
        t% = t% + 1
    ELSE
        IF v% = 1 THEN
            IF j% < 1 THEN ERROR 110
            j% = j% - 1
        END IF
        m% = m% - 1
        t% = t% - 1
        Restklasse v%, j%, m%, t%
    END IF
    IF z% = 1 THEN
        v1% = v%: j1% = j%: m1% = m%: t1% = t%
    ELSE
        v2% = v%: j2% = j%: m2% = m%: t2% = t%
    END IF
NEXT z%
END SUB 'ZeitSchieben _____'
```